# 豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

作成日 2017年5月23日 作成者 たま吉さん

# 目次

典 冊 禾 ゟ	_DACIC for Andri	20 CTM29 VA 99	リファレンスマニ	<b>7</b> II. DV/\(\) 1
<b>一四字</b> 》1 — 1	TDASIC for Aroun	no a i waa vu aa	'' ノ / '   ノ ノ ∧ ヾ =	_   / ///

目次.		i
1. 棋	概要	1
1.1	はじめに	1
1.2	2 システム構成	3
1.3	3 仕様	4
1.4	結線図	5
1.5	5 ボードピンの利用	6
2. 実	実行環境を用意しよう!	9
2.1	Blue Pill ボード・必要パーツの入手	9
2.2	2 開発環境 Arduino STM32 環境の構築	9
2.3	B 豊四季タイニーBASIC for Arduino STM32 の構築	9
2.4	豊四季タイニーBASIC for Arduino STM32 の動作確認	9
3. 復	使ってみよう!	10
3.1	入力可能状態は OK■(カーソル)	10
3.2	,	
3.3		
3.4	l PRINT の前に 10 を追加してプログラムにしよう	11
3.5	5 プログラムを実行してみよう	11
3.6	3 プログラムに追加する	12
3.7	7 行番号を並びなおす	12
3.8	3 やっぱり 10 行は不要、削除したい	13
3.9	) プログラムを保存したい	13
3.10	0 保存されたプログラムを読み込み	13
<b>4.</b> ■	画面操作	15
5.	プログラムの要素の説明(コマンド・関数等)	16
5.1	****	
5.2		
5.3	3 制御文	25
5.4	上 定数	25
5.5	5 文字列利用における制約	27
6. 名	各制御文の詳細	28
6.1	GOTO 指定行への分岐(制御命令)	28
6.2		
6.3	B RETURN GOSUB 呼び出し元への復帰(制御命令)	30
6.4	1 IF 条件判定(制御命令)	31
6.5	5 FORTO ~ NEXT 繰り返し実行(制御命令)	32

## 豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

6.6	END プログラムの終了 (制御命令)	33
7. 各:	コマンド・関数の詳細	34
7.1	LIST プログラムリストの表示(システムコマンド)	34
7.2	RENUM 行番号の振り直し(システムコマンド)	35
7.3	RUN プログラムの実行(システムコマンド)	36
7.4	LRUN 指定プログラム番号の実行(一般コマンド)	37
7.5	NEW プログラムの消去(システムコマンド)	38
7.6	DELETE プログラムの指定行の削除(システムコマンド)	39
7.7	SAVE 内部フラッシュメモリへのプログラム保存(システムコマンド)	40
7.8	LOAD 内部フラッシュメモリからプログラムを読み込む(システムコマンド)	41
7.9	FILES 内部フラッシュメモリ保存プログラムの一覧表示(システムコマンド)	42
7.10	ERASE 内部フラッシュメモリ上のプログラム削除(システムコマンド)	43
7.11	EXPORT 内部フラッシュメモリの内容のエクスポート(システムコマンド)	44
7.12	CONFIG システムの設定(システムコマンド)	46
7.13	SAVECONFIG システムの設定の保存(システムコマンド)	47
7.14	REM コメント(一般コマンド)	48
7.15	LET 変数に値を代入(一般コマンド)	49
7.16	FREE プログラム領域の残りバイト数の取得(数値関数)	50
7.17	PRINT 画面への文字表示(一般コマンド)	51
7.18	INPUT 数値の入力(一般コマンド)	53
7.19	CLS 画面表示内容の全消去(一般コマンド)	54
7.20	CLV 変数領域の初期化(一般コマンド)	55
7.21	COLOR 文字色の設定(一般コマンド)	56
7.22	ATTR 文字表示属性の設定(一般コマンド)	57
7.23	LOCATE カーソルの移動(一般コマンド)	58
7.24	VPEEK 画面指定位置の文字コード参照(数値関数)	59
7.25	ABS 絶対値の取得(数値関数)	60
7.26	MAP 数値のスケール変換(数値関数)	61
7.27	RND 乱数の発生(数値関数)	62
7.28	ASC 文字から文字コードへの変換(数値関数)	63
7.29	LEN 文字列の長さの取得(数値関数)	65
7.30	CHR\$ 文字コードから文字への変換(文字列関数)	66
7.31	STR\$ 変数が参照する文字列の取得・文字列の切り出し	67
7.32	BIN\$ 数値から 2 進数文字列への変換(文字列関数)	68
7.33	HEX\$ 数値から 16 進数文字列への変換(文字列関数)	69
7.34	DMP\$ 整数の小数付き数値文字列への変換(文字列関数)	70
7.35	REDRAW 画面表示の再表示(一般コマンド)	71
7.36	INKEY キー入力の読み取り (数値関数)	72

## 豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

7.37	DATE 現在時刻の表示(一般コマンド)	73
7.38	GETDATE 日付の取得(一般コマンド)	74
7.39	GETTIME 時刻の取得(一般コマンド)	75
7.40	SETDATE 時刻の設定(一般コマンド)	76
7.41	TICK 起動からの経過時間取得(数値関数)	77
7.42	RESETTICK 起動からの経過時間カウントのリセット(一般コマンド)	78
7.43	WAIT 時間待ち(一般コマンド)	79
7.44	GPIO GPIO 機能設定(一般コマンド)	80
7.45	OUT デジタル出力 (一般コマンド)	81
7.46	POUT PWM パルス出力(一般コマンド)	82
7.47	IN デジタル入力 (数値関数)	83
7.48	ANA アナログ入力(数値関数)	84
7.49	SHIFTIN デジタルシフトアウト入力 (数値関数)	85
7.50	SHIFTOUT デジタルシフトアウト出力 (一般コマンド)	87
7.51	SMODE シリアルポート機能切り替え (一般コマンド)	89
7.52	SOPEN シリアル通信オープン(一般コマンド)	91
7.53	SCLOSE シリアル通信クローズ(一般コマンド)	92
7.54	SPRINT シリアル通信文字列出力(一般コマンド)	93
7.55	SWRITE シリアル通信 1 バイト出力(一般コマンド)	95
7.56	SREADY シリアル通信データ受信データ確認(数値関数)	96
7.57	SREAD シリアル通信 1 バイト受信(数値関数)	98
7.58	I2CR I2C スレーブデバイスからのデータ受信(数値関数)	99
7.59	I2CW I2C スレーブデバイスへのデータ送信(数値関数)	101
7.60	PEEK 指定アドレスの値参照(数値関数)	103
7.61	POKE 指定アドレスへのデータ書き込み(一般コマンド)	104
7.62	EEPFORMAT 仮想 EEPROM のフォーマット(一般コマンド)	105
7.63	EEPWRITE 仮想 EEPROM のへのデータ書き込み(一般コマンド)	106
7.64	EEPREAD 仮想 EEPROM のからのデータ読み込み(数値関数)	107
7.65	PSET 点の描画(一般コマンド)	108
7.66	LINE 直線の描画(一般コマンド)	109
7.67	RECT 矩形の描画(一般コマンド)	110
7.68	CIRCLE 円の描画(一般コマンド)	111
7.69	BITMAP ビットマップ画像の描画(一般コマンド)	112
7.70	GPRINT 文字列の描画(一般コマンド)	113
7.71	CSCROLL キャラクタ画面スクロール(一般コマンド)	115
7.72	GSCROLL グラフィックスクロール(一般コマンド)	116
7.73	GPEEK 画面上の指定位置ピクセルの参照(数値関数)	118
7.74	GINP 指定領域のピクセルの有無判定(数値関数)	119

## 豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

,	7.75	TONE 単音出力(一般コマンド)	120
,	7.76	NOTONE 単音出力停止(一般コマンド)	121
9.	応用	月プログラム	122
9	9.1	I2C インタフェースの利用例	122
9	9.2	アナログ入力の利用例	122
9	9.3	SHIFTOUT コマンドを使った I/O 制御	122

## 1. 概要

#### 1.1 はじめに

「豊四季タイニーBASIC for Arduino STM32」は、Tetsuya Suzuki 氏が開発及び公開している「TOYOSHIKI Tiny BASIC for Arduino」を、Blue Pill ボード(STM32F103C8T6 搭載)に移植・機能拡張した整数型 Basic インタープリタ環境です。



写真 1 BluePill ボードにモニター、キーボード、圧電スピーカー、RTC 用バックアップ電池を接続した構成

オリジナル版のシンプルで分かりやすい環境を継承しつつ、TV モニターディスプレイと PS/2 キーボードを利用できる、インタープリタ環境に仕上がりました。



写真 2 起動直後の初期画面

本リファレンスマニュアルでは、オリジナル版の TinyBASIC の文法に加え、新しく追加した機能、コマンドについて解説・説明いたします。

本書の構成としては、前半に概要、システム構成、仕様、レガシーBASIC環境を始めて利用する方向けの解説、中盤にリファレンス、後半に応用の内容となります。


- ・「豊四季タイニーBASIC(オリジナル版)」に関する情報掲載サイト (作成中)
- ・「豊四季タイニーBASIC for Arduino STM32」に関する情報公開サイト (作成中)

## 著作権等

(作成中)

#### 1.2 システム構成

「豊四季 Tiny BASIC for Arduino STM32」では、次のような構成での利用を前提としています。



#### 1.3 仕様

## (1) ソフトウェア仕様

BASIC 仕様: 整数型 BASIC (符号付 16 ビット整数 -32767~32767)

BASIC プログラム用メモリ: 4096 バイト

変数: A~Zの26個(16ビット)、配列変数@(0)~@(99) 100個

ユーザー開放作業用メモリ: 2048 バイト

キャラクタ表示画面: 37×27 文字(6x8 フォント利用時)(999 バイト)

28×27 文字(8x8 フォント利用時) (756 バイト)

グラフィク表示画面: 224×216 ドット(6,048 バイト)

内部プログラム保存可能数: 10本(40,960バイト)

仮想 EEPROM 保存可能数: 254 ワード(うちシステム用の 4 ワード利用)

#### (2) ハードウェア仕様

NTSC ビデオ出力: NTSC-J(ビデオ信号 同期信号 0~1.0V)

PS/2 キーボード I/F: 5V 対応(5V トレラント ピン利用)

USB I/F: スケッチ書き込み、シリアル通信用

シリアル通信ポート: PA9、PA10 ピン利用(選択利用)


I2C I/F: PB6、PB7 ピン利用(占有)通信速度 400KHz

単音出力 : 3.3V 矩形波出力 1~32767Hz (デューティー比 50%)


GPIO: アナログ 12 ビット入力、デジタル入出力、PWM 出力

## 1.4 結線図


Blue Pill の場合の各パーツの取り付け・結線は次の図のようになります。



#### 1.5 ボードピンの利用



- 5V トレラント対応ピンでは、5V 系の信号入出力に接続可能です。
- コマンドや関数にてピンを指定する場合、ピン名称またはピン番号による指定が可能です。
- VBAT 端子は内蔵 RTC の外部電源によるバックアップに利用出来ます。



# ボート上のピン一覧

ピン名称	ピン番号	用途	説明
PA00	0	アナログ入力、デジタル IN/OUT	汎用
PA01	1	NTSC 同期信号	占有、他の利用禁止
PA02	2	アナログ入力、デジタル IN/OUT	汎用
PA03	3	アナログ入力、デジタル IN/OUT	汎用
PA04	4	アナログ入力、デジタル IN/OUT	汎用
PA05	5	アナログ入力、デジタル IN/OUT	汎用
PA06	6	アナログ入力、デジタル IN/OUT	汎用
PA07	7	NTSC 映像信号	占有、他の利用禁止
PA08	8	アナログ入力、デジタル IN/OUT、PWM	汎用
PA09	9	アナログ入力、デジタル IN/OUT、PWM、シリアル通信	汎用、シリアル通信(受信)
PA10	10	アナログ入力、デジタル IN/OUT、PWM、シリアル通信	汎用、シリアル通信(送信)
PA11	11	USB D-	占有、他の利用禁止
PA12	12	USB D+	占有、他の利用禁止
PA13	13	S-LINK、デジタル IN/OUT	S-LINK、汎用
PA14	14	S-LINK、デジタル IN/OUT	S-LINK、汎用
PA15	15	デジタル IN/OUT	汎用
PB00	16	アナログ入力、デジタル IN/OUT、PWM	汎用
PB01	17	アナログ入力、デジタル IN/OUT、PWM	汎用
PB03	19	デジタル IN/OUT	汎用
PB04	20	PS/2 キーボード I/F CLK	占有、他の利用禁止
PB05	21	PS/2 キーボード I/F DATA	占有、他の利用禁止
PB06	22	I2C SCL1	占有、他の利用禁止
PB07	23	I2C SDA1	占有、他の利用禁止
PB08	24	デジタル IN/OUT	
PB09	25	SOUND(PWM 出力)	占有、他の利用禁止
PB10	26	デジタル IN/OUT	汎用
PB11	27	デジタル IN/OUT	汎用
PB12	28	デジタル IN/OUT	汎用
PB13	29	デジタル IN/OUT	汎用
PB14	30	デジタル IN/OUT	汎用
PB15	31	デジタル IN/OUT	汎用
PC13	32	LED、デジタル IN/OUT ※プルアップ抵抗あり	
PC14	33	RTC へのクロック供給	占有、他の利用禁止
PC15	34	RTC へのクロック供給	占有、他の利用禁止

# ピン割り当て機能を利用するコマンド・関数一覧

ピン割り当て機能	機能利用コマンド・関数	説明
IN/OUT	GPIO	ピン機能割り当て
(デジタル入出力)	OUT	デジタル出力
	IN	デジタル入力
	SHIFTOUT	デジタルシフトアウト
	SHIFTIN	デジタルシフトイン
ADC	GPIO	ピン機能割り当て
(アナログ入力)	ANA0	アナログ入力
PWM	GPIO	ピン機能割り当て
(PWM 出力)	POUT	PWM 出力
I2C SDA	I2CW()	I2C 通信 データ送信
I2C SCL	I2CR()	I2C 通信 データ受信
SOUND 専用	TONE	サウンド出力
BOOKE 47/1	NOTONE	サウンド停止
LED	GPIO	ピン機能割り当て
шш	OUT	LED の点滅
RX, TX	SMODE	モード設定
(シリアル通信)	SOPEN	シリアル通信オープン
(フリアル巡旧)	SREAD()	1 バイト受信
	SWRITE	1 バイト送信
	SREADY()	データ受信可能チェック
	SPRINT	文字列送信
RTC 用クリスタル接続	DATE	
KIU 用ソリヘダル接枕	GETDATE	現在時刻表示
	GETTIME	日付の取得
	SETDATE	時刻の取得 時刻の設定
USB D+/USB D-		日初の設定   モード設定
USB D+/USB D- (USB シリアル通信)	SMODE SREAD()	T   T   T   T   T   T   T   T   T   T
	SWRITE	1 バイト送信
	SREADY()	1 ハ1 下区信   データ受信可能チェック
	SPRINT	
DG/0 + 1 1		文字列送信
PS/2 キーボード	CONFIG	US/日本語キーボード切り替え
DATA/CLK	SAVECONFIG	上記設定保存
NTSC 同期用・映像用	CONFIG	垂直同期信号調整
(ビデオ出力)	PRINT	文字列表示
	CLS	画面クリア
	REDRAW	画面の再表示
	CSCROLL	キャラクタ画面スクロール
	PSET	点の描画
	LINE	線の描画
	RECT	矩形の描画
	CIRCLE	円の描画
	BITMAP	ビットマップ画像の描画
	GPRINT	文字列の描画
	GSCROLL	グラフィックスクロール
	GPEEK	画面上のピクセル参照
	PINP	指定領域内のピクセル参照

## 2. 実行環境を用意しよう!

この章では、安価な Blue Pill ボードで「豊四季タイニーBASIC for Arduino STM32」を稼働させるための作業について説明します。

大まかな流れは次のようになります。

- 1. Blue Pill ボード・必要パーツの入手
- 2. 開発環境 Arduino STM32 環境の構築
  - 1) ソフトウェアの入手と設定
  - 2) Blue Pill ボードへのブートローダの書き込み
  - 3) Lチカスケッチ書き込み確認
- 3. 豊四季タイニーBASIC for Arduino STM32 の構築
  - 1) NTSC、PS/2 キーボードの接続
  - 2) ソフトウェア(スケッチ・ライブラリ)の入手と設定
  - 3) スケッチ書き込み
- 4. 豊四季タイニーBASIC for Arduino STM32 の動作確認
  - 1) NTSC ビデオモニターの表示確認
  - 2) PS/2 キーボードの動作動作確認
  - 3) USB・シリアル接続の動作確認

(以降作成中)

- 2.1 Blue Pill ボード・必要パーツの入手 (以降作成中)
- 2.2 開発環境 Arduino STM32 環境の構築

(以降作成中)

2.3 豊四季タイニーBASIC for Arduino STM32 の構築

(以降作成中)

2.4 豊四季タイニーBASIC for Arduino STM32 の動作確認

(以降作成中)

3. 使ってみよう!

この章では、「レガシーBASIC(古き良き時代の BASIC 言語)の雰囲気を知らない方々にその雰囲気をつかんでもらいましょう」ということで、軽いフットワークで解説進めます。

準備 OK? 前章の実行環境は完璧? それでは電源を入れて使ってみよう!

3.1 入力可能状態は OK■ (カーソル)

OK

この状態は、「いつでもこいや~」状態。 きみのコマンド入力を待っている状態。 何か打ち込んでみよう!

3.2 Hello, world を表示してみる

直接、次の文字を打ってみよう。 print "Hello,world" Enter

入力ミスの修正は[BS]キー、[DEL]キー、カーソルキーなどいつものキーが使えるよ。

入力出来たら、最後に[Enter]キーを押してみよう。

print "Hello,world" Enter
Hello,world
OK

表示出来たね。PRINT は文字列を表示するコマンドなのです!

入力は大文字、小文字どちらでも OK!

3.3 Hello,world を Hello,Tiny BASIC に変更してみる

カーソルキー[↑][↓][←][→]でカーソルを動かして、world の w に移動してみよう。 print "Hello, world" Hello, world

[DEL]キーで world を削除して、代わりに Tiny BASIC と入力しよう。

入力後に、[Enter]キーを押すのを忘れずに!

print "Hello, Tiny BASIC" Enter Hello, Tiny BASIC

OK

こんな感じで[Enter]がコマンド実行の合図だ。

[Enter]を入力する際、カーソルは行のどこにあっても構わない。

#### 3.4 PRINT の前に 10 を追加してプログラムにしよう

再びカーソル操作で print の前に 10 を挿入して[Enter]キーを押してみよう。

10 print "Hello,Tiny BASIC" Enter ■ello,Tiny BASIC OK

これでプログラムとして登録された。LIST コマンドで確認してみよう。

カーソルを OK の下の空き行に移動して、list と入力

10 print "Hello,Tiny BASIC" Enter Hello,Tiny BASIC
OK
list Enter
10 PRINT "Hello,Tiny BASIC"
OK

登録したプログラムの内容が表示されたね。

まだ、行番号10しか登録していないので、その1行だけの表示だ。

このように、先頭に数字(=行番号)をつけると行番号の後ろのコマンドがプログラムとして登録されるのだ。

## 3.5 プログラムを実行してみよう

う~ん、気分的に、画面を綺麗にしてから表示したくなった。

画面をきれいにしてから、3回実行してみよう。

cls Enter
OK
run Enter
Hello,Tiny BASIC
OK
run Enter
Hello,Tiny BASIC
OK
run Enter
Hello,Tiny BASIC
OK

3回同じ処理が実行出来たね。

CLS は画面を消すコマンド、RUN はプログラムを実行。よく使うコマンドだよ。

#### 3.6 プログラムに追加する

今の"画面きれいにしてから、3回表示"をプログラムにやらせてみよう! 次をように入力して、プログラムとして登録するよ。

5 cls Enter 7 for i=1 to 3 Enter 20 next i Enter

## 入力したプログラムを確認!

List Enter
5 CLS
7 FOR I=1 TO 3
10 PRINT "Hello, Tiny BASIC"
20 NEXT I
OK

今入力した 5,7,10 行が追加されたね。プログラムは行番号順に表示されるよ。

それでは実行! run コマンドで実行するよ。

run Enter
Hello,Tiny BASIC
Hello,Tiny BASIC
Hello,Tiny BASIC
OK

3 行表示出来た。

プログラムの実行順番は行番号順。CLS が最初に実行されるよ。 FOR 文は繰り返しを行う命令。3回10行の表示を行ったよ。

## 3.7 行番号を並びなおす

行番号の 5,7,10,20 と並びがちょっと美しくないですね。 直してみよう。

RENUM コマンドを使ってプロフラムの行番号を 10 ずつに整えるよ。

retum Enter OK List Enter 10 CLS 20 FOR I=1 TO 3 30 PRINT "Hello,Tiny BASIC" 40 NEXT I OK きれいに並んだね。 3.8 やっぱり 10 行は不要、削除したい

やっぱり、直前の表示内容は消したくない。

10 行はいらない。削除したい。行番号だけ入力して[Enter]を押してみよう。

10 Enter

これで10行が削除されるよ。確認してみよう。

List Enter
20 FOR I=1 TO 3
30 PRINT "Hello, Tiny BASIC"
40 NEXT I
OK

- 10 行が無くなったね。
- 20 行から始まるのは美しくない。行番号をまた振り直そう。

renum Enter
10 FOR I=1 TO 3
20 PRINT "Hello, Tiny BASIC"
30 NEXT I
OK

こんな感じ行番号だけの入力で、行削除。よく使う操作だよ。

3.9 プログラムを保存したい

今作成したプログラムは SRAM 上に保存されているのだ。 マイコンの電源を切ると消えてしまう・・・

消えないで欲しい・・・。そんな時、SAVE コマンドを使うよ。 SAVE コマンドはプログラムをフラッシュメモリに保存するよ。 フラッシュメモリに保存されたプログラムは電源を切っても消えないよ。 次のコマンドを打ってみましょう。

save Enter OK

これで保存できたはずだ。

3.10 保存されたプログラムを読み込み

本当に保存されているか確認してみましょう。

SRAM 上のプログラムを消してフラッシュメモリか読み込みと試してみよう。 NEW コマンドを打ってプログラムを初期化するよ。

new Enter

これで消えたはず。

LIST コマンドで確認。

list Enter OK 何も出てこない。本当に消えた。

次に LOAD コマンドでフラッシュメモリからプログラム読み込むよ。

loadEnter
OK
list Enter
10 FOR I=1 TO 3
20 PRINT "Hello,Tiny BASIC"
30 NEXT I
OK

成功! 復活出来た!・・・ これで基本的な操作は終了だ。

基本はバッチリ、準備万端! Tiny BASIC を使った更なるプログラミングをエンジョイしよう!

## 4. 画面操作

本章では画面操作について説明いたします。

以下に利用できるキーボード上のキーの一覧を示します。

TV(NTSC)出力・PS/2 キーボード利用環境できるキーです。

入力可能文字 : 半角英数字、半角記号

キーボード操作

カーソルの上下左右移動 : [←][→][↑][↓]

行の先頭に移動: [HOME]行の末端に移動: [END]

画面表示情報の先頭・末尾行に移動: [PageUP],[PageDown],

カーソルの前の文字の削除 : [BackSpace]

カーソル位置の文字の削除 : [Delete], [CTRL-X]

カーソル位置の下に 1 行空白追加 : [CTRL-N]

編集モードの切り替え

挿入モード・上書きボード切り替え : [Insert]

ローマ字カタカナ入力: [カタカタひらがな/ローマ字], [CTRL-K]

その他

入力行の確定 : [Enter]

画面の再表示: [F5]. [CTRL-R]実行プログラムの強制終了: [CTRL-C], [ESC]画面の表示内容消去: [F1], [CTRL-L]

テンキーの入力モード切替: NumLock英数字大文字・小文字切替: CapsLock

画面表示制御コマンド

画面の表示内容消去 : CLS

画面の文字色の設定: COLOR ※ (現時バージョンでは何もしない)画面の文字表示属性の設定: ATTR ※ (現時バージョンでは何もしない)

画面の再表示: REDRAW画面カーソル移動: LOCATE文字の表示: PRINT日付表示: DATE

# 5. プログラムの要素の説明(コマンド・関数等)

(ここにサンプルプログラムのソース)

(構成要素の説明)

(構成要素の一覧)

コマンド

行番号

コマンド区切り

数值•数值定数

変数

配列変数

代入文

式・演算子

文字列

関数 (数値関数、文字列関数)

制御文

(以降作成中)

#### 5.1 コマンド・関数

コマンドとは何らかの処理を行う命令文です。

コマンドにはプログラム作成や環境設定等に利用するシステムコマンドと、プログラム中で利用する一般コマンドがあります。

一般コマンドは、コマンドライン直接入力での実行のほか、プログラム中での利用も可能です。システムコマンドは、コマンドライン直接入力のみ利用できます。

システムコマンドはプログラム中での記述は可能ですが無視されます。

関数とは何らかの値を返す命令文です。関数には数値関数と文字列関数の2種類があります。

数値関数は数値を返す関数です。文字列関数は文字列を返す関数です。

数値関数はコマンドの引数や式のなどの数値として利用します。

文字列関数は PRINT, SPRINT, GPRINT, 文の引数にて利用できます。

関数単独での利用は出来ません。

以下に機能別に分類したコマンド及び関数の一覧を示します。

各コマンド及び関数の利用方法の詳細については、「7.各コマンド・関数の詳細」を参照して下さい。

#### ■ システムコマンド

LIST	プログラムリストの表示
LIST 開始行	開始行:表示を開始する行
LIST 開始行,終了行	終了行:表示を終了する行
RENUM	行番号の振り直し
RENUM 先頭行番号	先頭行番号:振りな直し開始先頭行
RENUM 先頭行番号,間隔	間隔:行間の増分
RUN	プログラムの実行
NEW	プログラムの消去
DELETE 行番号	プログラムの指定行の削除
DELETE 先頭行番号,末尾行番号	
SAVE	内部フラッシュメモリへのプログラム保存
SAVE [プログラム番号]	プログラム番号:保存先番号(0~9)
LOAD	内部フラッシュメモリからプログラムを読み込む
LOAD [プログラム番号]	プログラム番号:読み出し元番号(0~9)
FILES	内部フラッシュメモリの保存プログラムリスト表示
EXPORT	内部フラッシュメモリの内容をエクスポートする
EXPORT 対象番号	対象番号 エクスポートする番号
EXPORT 開始番号,終了番号	開始番号,終了番号 エクスポートする範囲
CONFIG 項目番号,設定値	システムの設定
SAVECONFIG	システム設定の保存
ERASE プログラム番号	内部フラッシュメモリ上のプログラム削除
ERASE 開始プログラム番号,終了プログラム番号	

# ■ プログラム関連

# コマンド

<b>REM</b> [コメント文]	コメント 省略形の'(シングルクオテーション)も可能
'[コメント文]	
LET 変数=式	変数・配列変数に値を代入(LET は省略可能)
LET @(添え字)=n1[,n2,n3,n4…]	配列変数では連続代入指定が可能
CLV	変数領域の初期化
LRUN プログラム番号	指定したプログラムをロードして実行する
LRUN プログラム番号,行番号	
LRUN プログラム番号,"ラベル"	

## 数値関数

ABS(整数)	絶対値の取得
MAP(値,開始範囲 1,終了範囲 1,開始範囲 2,終了範囲 2)	数値のスケール変換
ASC(変数)	変数が参照する文字列の取得 ・文字列の切り出し
ASC(変数,文字位置)	
ASC(文字列)	
ASC(文字列,文字位置)	
FREE()	プログラム領域の残りバイト数の取得
RND()	乱数の発生
INKEY()	キー入力の読み取り
LEN(変数)	文字列長の取得
LEN(文字列)	

## 文字列関数

CHR\$(文字コード)	画文字コードから文字への変換
BIN\$(数值[,桁指定))	数値から2進数文字列への変換
HEX\$(数值[,桁指定))	数値から 16 進数文字列への変換
DMP\$(数值)	整数の小数付き数値文字列への変換
DMP\$(数值,小数桁数)	小数桁数 0~4
DMP\$(数值,小数桁数,整数部桁数)	整数部桁数 0~8
STR\$(変数)	変数が参照する文字列の取得
STR\$(変数,先頭,長さ)	
STR\$(文字列)	
STR\$(文字列,先頭,長さ)	

# ■ 時間待ち・時間計測関連

# コマンド

RESETTICK	起動からの経過時間カウントのリセット
WAIT ミリ秒	時間待ち

# 数值関数

TICK()	起動からの経過時間取得
TICK([=-   ])	モード:0 下位読み出し 1:上位読み出し

## ■ 記憶領域操作関連

# コマンド

POKE アドレス,データ	指定アドレスへのデータ書き込み
数值関数	
PEEK(アドレス)	指定アドレスの値参照

## ■ キャラクタ表示関連

## コマンド

PRINT [#n,] 数値・文字列[; 数値・文字列…][;]	画面への文字表示 文末;で改行の抑制可能
INPUT 変数	数値の入力
INPUT 変数,オーバーフロー時の既定値	
INPUT プロンプト 変数	
INPUT プロンプト 変数, オーバーフロー時の既定値	
CLS	画面表示内容の全消去
COLOR 文字色,	文字色の設定 ※ターミナル版でのみ利用可能
COLOR 文字色,背景色	
ATTR 属性	文字表示属性設定 ※ターミナル版でのみ利用可能
LOCATE 横位置,縱位置	カーソルの移動
REDRAW	画面表示の再表示
CSCROLL x1,y1,x2,y2,方向	指定領域内での文字スクロール

## 数值関数

## ■ グラフィク表示関連

## コマンド

PSET x, y, 色	点の描画
LINE x1,y1,x2,y2,色	直線の描画
RECT x1,y1,x2,y2,色,モード	矩形の描画
CIRCLE x, y,半径,色,モード	円の描画
BITMAP x, y, アドレス, インデックス, 幅, 高さ [,倍率]	ビットマップ画像の表示
GPRINT X,Y,[#n,] 数値・文字列[;数値・文字列…][;]	指定位置に文字列描画
GSCROLL x1,y1,x2,y2,方向	指定領域内でのグラフィックスクロール

## 数値関数

GPEEK(横位置,縱位置)	画面上の指定位置ピクセルの参照
GINP(横位置,縦位置,高さ,幅,色)	指定領域のピクセルの有無判定

## ■ サウンド関連

## コマンド

TONE 周波数,出力期間	指定周波数のパスル出力
NOTONE	パルス出力の停止

## ■ RTC(時刻)関連

# コマンド

DATE	現在時刻の表示
GETDATE 変数 1,変数 2,変数 3,変数 4	日付の取得 変数1ら順に年,月,日,曜日コードを格納
GETTIME 変数 1,変数 2,変数 3	時刻の取得 変数1から順に時,分,秒を格納
SETDATE 年,月,日,時,分,秒	時刻の設定

## ■ GPIO・入出力関連

## コマンド

GPIO ピン番号,機能名	GPIO 機能設定
OUT ピン番号,出力値	デジタル出力
POUT ピン番号, デューティー値	PWM パルス出力
POUT ピン番号, デューティー値,周波数	
SHIFTOUT データピン,クロックピン, 出力形式, 出力データ	デジタルシフトアウト出力

## 数值関数

I2CR(デバイスアドレス,コマンドアドレス,コマンド長,データア	I2C スレーブデバイスからのデータ受信
ドレス,データ長)	
I2CW(デバイスアドレス,コマンドアドレス,コマンド長,データア	I2C スレーブデバイスへのデータ送信
ドレス,データ長)	
IN(ピン番号)	デジタル入力
SHIFTIN(データピン番号,クロックピン番号,入力形式)	デジタルシフトイン入力
SHIFTIN(データピン番号,クロックピン番号,入力形式,条件)	

## ■ シリアル通信関連

## コマンド

SMODE モード	GPIO シリアルポート、USB シリアルポート機能切り替え
SMODE モード,"通信速度"	
SOPEN "通信速度"	シリアルポートのオープン
SCLOSE	シリアルポートのクローズ
SPRINT[#n,] 数値・文字列; 数値・文字列;	シリアルポートへの文字列出力
SWRITE データ	シリアルポートからの 1 バイト送信

# 数值関数

SREADY()	シリアル通信 受信データ確認
SREAD()	シリア通信 1バイト受信

## ■ 仮想 EEPROM 関連

## コマンド

EEPFORMAT	仮想 EEPROM のフォーマット
EEPWRITE アドレス,データ	仮想 EEPROM のへのデータ書き込み

## 数値関数

EEPREAD(アドレス)	仮想 EEPROM のからのデータ読み込み

システムコマンド、一般コマンドの詳細については「7各コマンド・関数」を参照して下さい。

#### 5.2 数値・文字・文字列・変数・演算子・式・文

#### ■ 数値

「Tiny BASIC for Arduino STM32」で使える数値は整数型のみとなります。

また整数型は 16 ビット幅、有効範囲は-32767~32767 となります。式、数値定数、数値関数、変数、配列変数はすべてこれに従います。

数値の表記は次の形式が可能です。

10 進数表記(-32767~32767) : -1、-32767, 100

16 進数表記(1 桁~4 桁) : \$1345, \$abcd, \$Abcd

## ■ 文字・文字列

「Tiny BASIC for Arduino STM32」では 1 バイト半角英数字の文字列を扱うことが出来ます。 文字列の表記は次の通りです。

"文字列": ダブルクォーテーションで囲みます。

文字列は1~255文字まで指定可能です。

#### ■ 変数

変数とは数値を格納する保存領域です。数値と同様に式に利用できます。

格納する数値は 16 ビット幅、有効範囲は-32767~32767 となります。

変数は数値型のみ利用可能ですが、文字列の格納アドレスを代入することにより、文字列の参照を行うことが出来出来ます。

変数は通常の変数の他に配列変数があります

#### 変数の表記

通常の変数:変数名 A~Zの1文字で表記、26 個利用可能

例) A=4

Z=Z+1

A="ABCD"

配列変数 : @(添え字)の形式で添え字は数値で 0~99(@(0)~@(99))まで利用可能

添え字の数値には式、変数等の利用が可能

例) @(0)=30/5,

@(A+1)=5

代入式において、指定した添え字を起点として連続代入が可能

例)@(10)=100,200,300,400,500

## ■ 演算子

数値演算で利用できる演算子を示します。

記述例のA,Bには数値、変数、配列変数、関数、カッコで囲んだ式が利用できます。

## 算術演算子

演算子	説明	記述例	
+	足し算	A+B	A と B を足す
-	引き算	A-B	A から B を引く
*	掛け算	A*B	A と B の積
/	割り算	A/B	A を B で割る
%	剰余算	A%B	AをBで割った余り

## ビット演算子

演算子	説明	記述例
&	ビット毎の AND 演算	A&B A と B のビット毎の AND
	ビット毎の OR 演算	A B AとBのビット毎の OR
>>	ビット右シフト演算	A>>B A を右に B ビットシフト
<<	ビット左シフト演算	A< <b a="" b="" td="" を左に="" ビットシフト<=""></b>
~	ビット毎の反転	~A Aの各ビットを反転
^	ビット毎の XOR	A^B AとBのXOR

比較演算、論理演算の論理反転は、0が偽、0以外が真となります。

0以外が真となりますので、1、-1はとも真となります。

## 比較演算子

演算子	説明	記述例		
=	等しいかを判定	A=B	AとBが等しければ真,異なれば偽	
! =	異なるかを判定	A<>B	A と B が異なれば真,等しければ偽	
<	小さいかを判定	A <b< td=""><td>A が B 未満であれば真、そうでなければ偽</td></b<>	A が B 未満であれば真、そうでなければ偽	
<=	小さいまたは等しいかを判定	A<=B	A が B 以下であれば真、そうでなければ偽	
>	大きいかを判定	A>B	AがBより大きければ真、そうでなければ偽	
>=	大きいまたは等しいかを判定	A>=B	A が B 以上であれば真、そうでなければ偽	

## 論理演算子

演算子	説明	記述例		
AND	論理積	A AND B A.Bが真なら真、でなければ偽		
OR	論理和	A OR B A,B どちらかが真なら真、でなければ偽		
!	否定	!A Aが真なら偽、偽なら真		

## ■ 演算子の優先順序

演算子の優先度を下記に示します。優先度の数値が小さいほど優先度が高くなります。 計算結果が意図した結果にならない場合は、優先度の確認、括弧をつけて優先度を上げる等の対応を 行って下さい。

## 演算子の優先度

優先度	演算子
1	括弧で囲った式
2	!, ~
3	*, / , % , &,   . << , >>,^
4	+ , - ,
5	=, <> , != , >, >= , < , <= , AND , OR

比較演算子と論理演算は優先度が同レベルです。比較演算子の演算を先に行う場合は、 括弧を使って優先度を上げて下さい。

?1<5 and 2>3

0

?(1<5) and (2>3)

1

#### ■ 式

式とは 1、1+1、A、A+1、ABS(-1) などの演算・値の評価を伴う記述をいいます。 式はプログラム実行にて計算・評価が行われ、 1 つの整数値の値として振る舞います。 変数への値の代入、コマンド、条件判定(IF 文)、関数に引数に式が用いられた場合は、評価後の値が コマンドおよび関数に渡されます。

#### 式の利用:

変数への値の代入(代入命令 LET は省略可能) LET A=(1+10)\*10/2 A=5\*5+3\*2 @(I+J) = I\*J

コマンド・関数の引数 LOCATE X+I\*2, J:PRINT "\*" OUT PC13, I>J A=EEPREAD(I+J+1)/2

#### ■文

文とはコマンド、式、関数、コメント、ラベルを組み合わせて記述した命令です。 コマンドはコロン:を使って 1 行に複数記述することができます。 10 I=I+1:LOCATE 0,I:COLOR 7,0:PRINT "Hello":GOTO 50

#### コメント文

文において、REM、および省略形のシングルクオーテーション'を使った以降の文はコメントとなります。

例:

10 'Sample program

20 REM Sample program

30 PRINT "Hello": 'print hello

#### ラベル

行の先頭のダブルクオテーションで囲った文字列はラベルとして動作します。 ラベル自体は実行時に何も行いません。GOTO 文、GOSSUB 文のジャンプ先の指定で利用しま す。ラベルの後ろにはコマンドを続けて記述することができます。

10 "LOOP": PRINT "Hello"

30 GOTO "LOOP"

#### 5.3 制御文

制御文は制御命令を使った文です。複数のキーワード、式の組み合わせで構成されます。

FOR 文の例:

FOR A=0 TO 10 STEP 5

PRINT A

NEXT A

プログラムはRUNコマンドを実行することにより、先頭行から行番号順に逐次実行されます。 この逐次実行の流れは、制御文を用いることで条件分岐、繰り返しを行うことが出来ます。 この制御文としては、次の命令があります。

制御命令	説明
GOTO 行番号	指定行へのジャンプ
GOTO ラベル	
GOSUB 行番号	サブルーチンの呼び出し
GOSUB ラベル	
RETURN	GOSUB 文サブルーチン呼び出しからの復帰
IF 条件式 実行文 1	条件判定
IF 条件式 実行文 1 ELSE 実行文 2	
FOR 変数=初期値 TO 最終値 実行文(複数行可能) NEXT 変数	繰り返し
FOR 変数=初期値 TO 最終値 STEP 増分 繰り返し実行文 NEXT 変数	
END	プログラムを終了する
LRUN プログラム番号	指定したプログラムをロードして実行する
LRUN プログラム番号,行番号	
LRUN プログラム番号,"ラベル"	

#### 5.4 定数

「Tiny BASIC for Arduino STM32」では次の定数が利用出来ます。 定数はコマンドの引数や式の中で数値関数と同等に利用出来ます。

## ■ 1ビット入出力値

HIGH, LOW

HIGH は 1,LOW は 0 が割り当てられています。

各コマンドの引数、IF文、式にて利用出来ます。

## ■ メモリ領域先頭アドレス定数

VRAM, VAR, ARRAY, PRG, MEM, FNT

データ領域を参照するための定数です。各定数の詳細は次の通りです。

定数名	用途	領域サイズ	定数值
VRAM	画面表示用メモリ(CW×CH)	2048	\$0000
VAR	変数領域(A~Z)	56	\$0800
ARRAY	配列変数領域(@(0)~@(99))	200	\$0838

豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

PRG	プログラム領域	4096	\$0900
MEM	ユーザーワーク領域	2048	\$1900
FNT	256 文字分(フラッシュメモリ)	2048	\$2100

定数値のアドレスは仮想的なアドレスです。実アドレスとは異なります。

PEEK、POKE、I2CW、I2CR、BITMAP等のメモリ領域を利用するコマンドで利用できます。

## ■ 画面表示の定数

CW: キャラクタ画面の横文字数(6x8 フォントの場合 37) CH: キャラクタ画面の縦行数(6x8 フォントの場合 27)

GW : グラフィック画面の横ドット数(224)GH : グラフィック画面の縦ドット数(216)

画面サイズを参照する定数です。

## ■ 方向を示す定数

UP : 値 0 DOWN : 値 1 RIGHT : 値 2 LEFT : 値 3

CSCROLL, GSCROLL コマンドでスクロール方向を指定に利用出来ます。

#### ■ ピン番号定数

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08 PA09, PA10, PA11, PA12, PA13, PA14, PA15 PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08 PB09, PB10, PB11, PB12, PB13, PB14, PB15 PC13, PC14, PC15

GPIO、OUT、IN、ANA、SHIFTOUT、SHIFTIN コマンド・関数のピン番号の指定に利用します。 各ピン番号定数に実際のピン番号 0 ~ 34 が割り当てられています。

#### ■ GPIO モード設定定数

OUTPUT\_OD, OUTPUT, INPUT\_PU, INPUT\_PD, ANALOG, INPUT\_FL GPIO コマンドのモード設定を行うための定数です。

#### ■ ビット方向定数

LSB, MSB

ビット送信等で上位、下位を指定するための定数です。

SHIFTIN、SHFITOUT コマンドで利用します。

#### 5.5 文字列利用における制約

「Tiny BASIC for Arduino STM32」では制約付きで文字列の利用をサポートしています。 変数への代入も可能です。

```
例:
10 A="ABCDE"
20 PRINT A
RUN
ABCDE
OK
```

変数への文字列代入は、文字列格納アドレスを変数に代入することにより実現しています。

C言語のポインタによる参照方式と同等です。

ただし、「Tiny BASIC for Arduino STM32」では文字列情報を[長さ+文字列]の形式で管理しているため、文字列の参照を代入した変数の値を変更して利用することは行えません。

下記のような利用は出来ません。行った場合、意図しない動作となります。

例:

10 A="ABCDE"

20 A=A+1

30 PRINT A

また、コマンドラインでの利用は行えません。文字列参照が正しく行うことが出来ません。

例:

A="ABCDE" OK

?A

変数に代入した文字列の操作を行う次の関数を用意しています。

・文字列の長さの取得 : LEN()

10 A="ABCDE" 20 L=LEN(A) RUN OK

・変数で参照している文字列の取得、部分切り出し:STR\$()

```
10 A="ABCDE"
20 PRINT STR$(A)
30 PRINT STR$(A,4,1)
RUN
ABCDE
D
OK
```

・指定位置の文字コードの取得 :ASC()

```
10 A="ABCDE"
20 C=ASC(A,4,1)
```

RUN

68

OK

#### 6. 各制御文の詳細

#### 6.1 GOTO 指定行への分岐(制御命令)

## ■ 書式

GOTO 行番号 GOTO "ラベル"

## ■ 説明

プログラムの実行を行番号またはラベルで指定した行に移ります。

行番号には数値、式(変数、関数を含む)が利用できます。

ラベルを指定した場合は、行先頭に同じラベルがある行に移ります。

ラベルはダブルクオテーション(")で囲って指定します。

10 I=0

20 "LOOP"

30 PRINT "@"

40 I=I+1:IF I=5 GOTO 60

50 GOTO "LOOP"

60 END

上記の例では、40 行の GOTO で 60 行に移動、50 行のラベル指定で 20 行に移動しています。 ラベルを使うとプログラムの流れがわかりやすくなります。

ラベルの使用は行番号に比べるとラベル文字列検索の分、若干処理に時間がかかります。 行先頭以外にラベルがある場合は、GOTO 文、GOSSUB 文のジャンプ先として参照はされません。

## ■ エラーメッセージ

Undefined line number or label : 条件

Overflow: 指定した数値が-32767 ~ 32767 を超えている

Syntax error : 文法エラー、書式と異なる利用を行った

## ■ 利用例

100 GOSUB "SUB01"

110 GOSUB "SUB02"

120 N=0

130 "LOOP"

140 PRINT "N=";N

150 N=N+1:IF N<5 GOTO "LOOP"

160 END

170 "SUB01"

180 PRINT "SUB01"

190 RETURN

200 "SUB02"

210 PRINT "SUB02"

220 RETURN

#### 6.2 GOSUB サブルーチンの呼び出し(制御命令)

#### ■ 書式

GOSUB 行番号

GOSUB "ラベル"

#### ■ 説明

プログラムの実行を一旦、行番号またはラベルで指定した行に移ります。

移った先で RETURN 命令により、呼び出した GOSUB 命令の次の位置に戻ります。

行番号には数値、式(変数、関数を含む)が利用できます。

ラベルを指定した場合は、行先頭に同じラベルがある行に移ります。

ラベルはダブルクオテーション(")で囲って指定します。

- 10 GOSUB "PRN\_LOGO"
- 20 GOSUB "PRN DATE"
- 30 GOSUB 200
- 40 END
- 50 "PRN LOGO"
- 60 PRINT "Tiny BASIC for Arduino STM32"
- 70 RETURN
- 80 "PTN DATE"
- 90 PRINT "Edition V0.3 2017/04/01"
- 100 RETURN
- 200 PRINT "Ready"
- 210 RETURN

上記の例では、10行、20行でラベルを指定してサブルーチンを呼び出しています。

30 行では、行番号を指定してサブルーチンを呼び出しています。

ラベルを使うことで、プログラムの実行がわかりやすくなります。

ラベルの使用は行番号に比べるとラベル文字列検索の分、若干処理に時間がかかります。

行先頭以外にラベルがある場合は、GOTO文、GOSSUB文のジャンプ先として参照はされません。

#### ■ エラーメッセージ

Undefined line number or label : 条件

Overflow : 指定した数値が-32767 ~ 32767 を超えている

GOSUB too many nested : GOSUB のネスト数が規定を超えた Syntax error : 文法エラー、書式と異なる利用を行った

## ■ 利用例

- 100 GOSUB "SUB01"
- 110 GOSUB "SUB02"
- 120 N=0
- 130 "LOOP"
- 140 PRINT "N=";N
- 150 N=N+1:IF N<5 GOTO "LOOP"
- 160 END
- 170 "SUB01":PRINT "SUB01":RETURN
- 200 "SUB02":PRINT "SUB02":RETURN

#### 6.3 RETURN GOSUB 呼び出し元への復帰(制御命令)

#### ■ 書式

**RETURN** 

## ■ 説明

直前に呼び出された GOSUB の次の処理に復帰します。

詳細は「6.2 GOSUB サブルーチンの呼び出し(制御命令)」を参照して下さい。

## ■ エラーメッセージ

Undefined line number or label : 条件

Overflow : 指定した数値が-32767 ~ 32767 を超えている RETURN stack underflow : GOSUB の呼び出しがないのに RETURN を実行

Syntax error : 文法エラー、書式と異なる利用を行った

## ■ 利用例

- 100 GOSUB "SUB01"
- 110 GOSUB "SUB02"
- 120 N=0
- 130 "LOOP"
- 140 PRINT "N=";N
- 150 N=N+1:IF N<5 GOTO "LOOP"
- 160 END
- 170 "SUB01"
- 180 PRINT "SUB01"
- 190 RETURN
- 200 "SUB02"
- 210 PRINT "SUB02"
- 220 RETURN

### 6.4 IF 条件判定(制御命令)

#### ■ 書式

IF 条件式 実行文 1

IF 条件式 実行文 1 ELSE 実行文 2

### ■ 説明

条件式の真偽判定を行い、真の場合は実行文1を実行します。

偽の場合 ELSE 文の指定があれば実行文 2 を実行します。なければ次の行にスキップします。 真偽は次の条件となります。

真:値が0以外の数値

偽:値が0

真偽判定を行う、条件式には次の指定が可能です。

式: 例 IFA>5B=B-1

IF A+1 PRINT A

IF A & 1 TONE 440,200

IF INKEY() = ASC("A") GOTO 100

 定数
 : 例 IF HIGH GOTO 100

 数值
 : 例 IF 123 PRINT "123"

関数 : 例 IF RND(1) X=X+1

実行文には IF、GOTO、GOSUB 等の制御命令を使うことも可能です。

### (注意) ELSE 利用の制約(ぶら下がり ELSE に関する補足)

IF 文をネストして利用した場合、ELSE 文は直前の IF 文に対応します。

例: IF A=1 IF B=1? "A,B=1" ELSE? "A=1,B<>1"

上記の ELSE は 2 番目の IF 文に対応します。

この「ELSE 文は直前の IF 文に対応」の条件で、ELSE 文に IF 文をネスト出来ます。

例: IF A=1 IF B=1 ? "A,B=1" ELSE IF B=2 ? "A=1,B=2" ELSE IF B=3 ?"A=1,B=3" 上記では、A=1,B=1,2,3の条件に対して対応する表示メッセージを表示する例です。

### エラーメッセージ

IF without condition : 条件

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

10 A=INKEY()

20 IF A=ASC("Y") PRINT

#### 6.5 FOR TO ~ NEXT 繰り返し実行(制御命令)

#### ■ 書式

FOR 変数=初期値 TO 最終値 実行文(複数行可能) NEXT 変数 FOR 変数=初期値 TO 最終値 STEP 増分 繰り返し実行文(複数行可能) NEXT 変数

#### ■ 説明

変数を初期値から最終値まで増やし、FOR から NEXT の間の実行文を繰り返し実行します。 変数が最終値に達した時点で繰り返しを止め、NEXTの次の命令の実行を行います。

STEP にて増分を指定しない場合、増分は1となります。

STEP を用いた場合は、マイナス値を含め任意の増分の指定が可能です。

10 PRINT "start."

20 FOR I=0 TO 5 STEP 2

30 PRINT I

40 NEXT I

50 PRINT "done."

#### 実行結果

start

0

2

4

done.

上記の例では I を 0 から 5 まで、2 ずつ増加して 30 行の命令を繰り返し実行します。 Iが4の時、増分2を足すと終了値5を超えた6となるので繰り返しを終了し、50行を実行します。 FOR 文はネストも可能です。

### ■ エラーメッセージ

FOR without variable : FOR 文で変数を指定していない : FOR 文で TO を指定していなし FOR without TO NEXT without counter FOR too many nested : NEXT に対応する FOR 文が無い

: FOR 文のネストが多すぎる

: 文法エラー、書式と異なる利用を行った Syntax error

Overflow : 指定した数値が-32767 ~ 32767 を超えている

## ■ 利用例

画面の指定位置に\*を表示する(FOR 文のネストの例)

10 CLS

20 FOR Y=5 TO 10

30 FOR X=5 TO 10

40 LOCATE X,Y:PRINT "\*"

50 NEXT X

60 NEXT Y

## 6.6 END プログラムの終了(制御命令)

## ■ 書式

END

## ■ 説明

プログラムの実行を終了します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

# ■ 利用例

特定の条件でプログラムを終了する

10 I=0

20 PRINT I

30 IF I=5 END

40 GOTO 20

## 7. 各コマンド・関数の詳細

7.1 LIST プログラムリストの表示(システムコマンド)

#### ■ 書式

LIST

LIST 表示開始行番号

LIST 表示開始行番号,表示終了行番号

### ■ 引数

表示開始行番号:表示を開始する行番号(1 ~ 32767) 表示終了行番号:表示を終了する行番号(1 ~ 32767)

## ■ 説明

プログラムリストの表示を行います。

引数を指定しない場合は、全てのプログラムを表示します。

表示開始番号を指定した場合は、その番号以降のプログラムリストを表示します。

表示開始番号、表示終了番号を指定した場合は、そこ範囲のプログラムリストを表示します。

表示したプロフラムリストは、カーソルを移動して編集することが出来ます。

編集後は必ず[ENTER]キーを押して入力確定を行って下さい。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

```
LIST
10 I=0
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
OK

list 20,30
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
OK
```

### 7.2 RENUM 行番号の振り直し(システムコマンド)

### 走書 ■

RENUM

RENUM 開始行番号

RENUM 開始行番号, 增分

### ■ 引数

開始番号: 振り直しをする新しい行番号の開始番号 (1 ~ 32767)

増 分: 行番号の増分(1 ~ 32767)

#### ■ 説明

プログラム全体の行番号を指定した条件にて振り直します。

行番号を指定した開始番号から指定した増分で振り直します。

行番号は1~32767の範囲まで振ることが出来ます。

引数を省略した場合は、行番号を10行から10間隔で振り直します。

開始番号だけを指定した場合は、指定した開始番号から 10 間隔再振り直します。

開始番号と増分を指定した場合は、指定した開始番号から指定した増分で振り直します。

振り直しにおいて、GOTO文、GOSUB文のとび先の行番号も正しく更新されます。

開始行番号には変数、式の指定は行えません。必ず数値を指定して下さい。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

Illegal value : 振り直しをする新しい行番号が有効範囲を超えている

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

```
LIST
```

10 I=0

20 PRINT "\*";

30 I=I+1:IF I<10 GOTO 20

40 PRINT

50 END

OK

RENUM 100,10

OK

LIST

100 I=0

110 PRINT "\*";

120 I=I+1:IF I<10 GOTO 110

130 PRINT

140 END

7.3 RUN プログラムの実行(システムコマンド)

## ■ 書式

RUN

### ■ 引数

なし

### ■ 説明

プログラムの実行を行います。

実行中のプログラムを強制終了するには[ESC]キーを 2 回連続して押すか、[CTRL+C]キーを押して下さい。WAIT 命令で長い時間待ちを行っている場合は、強制終了に時間がかかる場合があります。

実行中は、カーソルが非表示となります。

プログラム実行中にエラーが発生した場合は実行を終了します。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

### ■ 利用例

LIST

10 I=0

20 PRINT "\*";

30 I=I+1:IF I<10 GOTO 20

40 PRINT

50 END

OK

RUN

\*\*\*\*\*\*

#### 7.4 LRUN 指定プログラム番号の実行(一般コマンド)

### ■ 書式

LRUN プログラム番号

LRUN プログラム番号,行番号

LRUN プログラム番号、"ラベル"

### ■ 引数

プログラム番号 実行するプログラムの番号 0 ~ 9 行番号 プログラムの実行開始行番号 1~32767 "ラベル" プログラムの実行開始行のラベル指定

#### ■ 説明

プログラム番号で指定したプログラムをフラッシュメモリからロードし、実行します。 引数に行番号またはラベルの指定がある場合は、その行からプログラムの実行を開始します。

LRUN にてプログラムを実行し場合、変数領域の初期化は行われません。 直前のプログラムが設定した変数の値を引き継ぎます。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : プログラム番号の指定が 0~9 の範囲外である

Program not found : 指定したプログラム番号にプログラムが保存されていない

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

入力したプログラム番号を実行する。

10 INPUT N

20 LRUN N

## 7.5 NEW プログラムの消去 (システムコマンド)

## ■ 書式

NEW

## ■ 引数

なし

## ■ 説明

プログラム領域のプログラム、変数、配列変数を消去します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

## ■ 利用例

```
LIST
10 I=0
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
OK

NEW
OK
LIST
```

### 7.6 DELETE プログラムの指定行の削除(システムコマンド)

### ■ 書式

DELETE 行番号

DELETE 先頭行番号,末尾行番号

## ■ 引数

行番号 : 削除対象の行番号 1~32767先頭番号 : 削除対象範囲の先頭番号 1~32767末尾番号 : 削除対象範囲の末尾番号 1~32767

## ■ 説明

プログラムの指定した行を削除します。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

Illegal value : 先頭番号と末尾番号の指定に矛盾がある

### ■ 利用例

**DELETE 20,50** 

### 7.7 SAVE 内部フラッシュメモリへのプログラム保存(システムコマンド)

#### ■ 書式

SAVE

SAVE プログラム番号

## ■ 引数

プログラム番号: 0~9

#### ■ 説明

プログラムをマイコン内のフラッシュメモリに保存します。

プログラムは最大で9つ保存可能です。保存先はプログラム番号0~9で指定します。

プログラム番号を指定しない場合、プログラム番号に保存します。

プログラム番号の指定には、変数や式の指定は出来ません。

(注意) NTSC 表示版では SAVE コマンド実行時に画面表示が乱れる場合があります。

(注意) SAVE コマンドはプログラム内での実行は出来ません。実行しても無視されます。 直接コマンド入力を行った場合のみ有効です。

#### ■ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Illegal value : プログラム番号の指定が 0~9 の範囲外である Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

プログラム番号 1 にプログラムを保存する SAVE 1 OK 7.8 LOAD 内部フラッシュメモリからプログラムを読み込む(システムコマンド)

### ■ 書式

LOAD

LOAD プログラム番号

### ■ 引数

プログラム番号: 0~9

#### ■ 説明

マイコン内のフラッシュメモリからプログラムを読み込みます。 読み込み前のプログラム、変数、配列変数は消去されます。

プログラム番号 0~9 で指定します。プログラムは最大で 10 つ保存可能です。 プログラム番号の指定しない場合、プログラム番号 0 を読み込みます。 プログラム番号の指定には、変数や式の指定は出来ません。

(注意) LOAD コマンドはプログラム内での実行は出来ません。実行しても無視されます。 直接コマンド入力を行った場合のみ有効です。

#### ■ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Illegal value : プログラム番号の指定が 0~9 の範囲外である

Program not found : 指定したプログラム番号にプログラムが保存されていない

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

プログラム番号 1 を読み込む LOAD 1 OK 7.9 FILES 内部フラッシュメモリ保存プログラムの一覧表示(システムコマンド)

#### ■ 書式

**FILES** 

#### ■ 引数

なし

#### ■ 説明

マイコン内のフラッシュメモリに保存されているプログラム(プログラム番号 0~9)の先頭行をリスト表示します。プログラム番号にプログラムが保存されていない場合は(none)と表示されます。

プログラム先頭行にコメントをつけると、一覧表示でのプログラムの内容が分かり易くなります。

- (注意) FILES コマンドはプログラム内での実行は出来ません。実行しても無視されます。 直接コマンド入力を行った場合のみ有効です。
- (注意) 豊四季 Tiny BASIC (ファームウェア) の新規利用または更新を行った直後は、フラッシュメモリ上の既存データの内容の不整合により正しく表示できない場合があります。その場合は、ERASE コマンドにてフラッシュメモリ上のプログラムの消去を行って下さい。

### ■ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

内部フラッシュメモリに保存されているプログラム一覧の表示

files

0:1' Edit bitmap

1:1'RTC TEST

2:(none)

3:(none)

4:(none)

5:(none)

6:(none)

7: (none)

8:(none)

9:1'LED Blink

7.10 ERASE 内部フラッシュメモリ上のプログラム削除(システムコマンド)

#### ■ 書式

ERASE プログラム番号

ERASE 開始プログラム番号、終了プログラム番号

### ■ 引数

プログラム番号 開始プログラム番号 終了プログラム番号

## ■ 説明

マイコン内のフラッシュメモリに保存されているプログラム(プログラム番号 0~9)の削除を行います。削除対象プログラムは、コマンド引数にてプログラム番号を指定するか、削除するプログラム番号の範囲 (開始プログラム番号、終了プログラム番号) にて指定します。

(注意) ERASE コマンドはプログラム内での実行は出来ません。実行しても無視されます。 直接コマンド入力を行った場合のみ有効です。

### ■ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

フラッシュメモリに保存されている全てのプログラムの削除を行う。

ERACE 0,9

OK

**FILES** 

0:(none)

1:(none)

2:(none)

3:(none)

4:(none)

5:(none)

6:(none)

7:(none)

8:(none)

9:(none)

7.11 EXPORT 内部フラッシュメモリの内容のエクスポート(システムコマンド)

#### ■ 書式

**EXPORT** 

EXPORT 対象番号

EXPORT 開始番号、終了番号

#### ■ 引数

対象番号 : エクスポートするプログラム保存番号 0 ~ 9

開始番号,終了番号 : エクスポートするプログラム保存番号の範囲 0 ~ 9

### ■ 説明

マイコン内のフラッシュメモリに保存されているプログラム(プログラム番号 0~9)を出力表示します。ターミナルソフト等を利用することにより、出力表示されたプログラムリストをコピーすることで、内部保存されたプログラムのバックアップを行うことが出来ます。

引数に何も指定しない場合は、0 ~ 9 までの全てのプログラムを出力します。 対象番号を指定した場合は、該当するプログラム番号の内容のみ出力します。 開始番号と終了番号を指定した場合はその範囲のプログラムを出力します。

出力形式は次の形式となります。

NEW

10 'oscilloscope

20 CLS

30 GPIO PB01, ANALOG

40 "LOOP"

50 R=ANA(PB01)

60 Q=R/20

70 LOCATE 0,0:?R;"

80 PSET GW-1,GH-Q-4,1

90 WAIT 50

100 GSCROLL 8,208,3

110 GOTO "LOOP"

SAVE 1

プログラム番号毎のプログラムリストの先頭に"NEW"コマンド、末尾に"SAVE プログラム番号"が付加されます。ターミナルソフトにペーストすることにより、フラッシュメモリ内に再登録することが出来ます。

(注意) EXPORT コマンドはプログラム内での実行は出来ません。実行しても無視されます。 直接コマンド入力を行った場合のみ有効です。

## ■ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

プログラム番号 1,2 に保存されているプログラムのエクスポート

```
EXPORT 1,2
NEW
10 'oscilloscope
20 CLS
30 GPIO PB01, ANALOG
40 "LOOP"
50 R=ANA(PB01)
60 Q=R/20
70 LOCATE 0,0:?R;" "
80 PSET GW-1,GH-Q-4,1
90 WAIT 50
100 GSCROLL 8,208,3
110 GOTO "LOOP"
SAVE 1
NEW
10 CLS
20 RECT 0,0,GW,GH,1,1
30 RESETTICK
```

40 FOR I=0 TO 215

50 FOR X=0 TO 200 STEP 16

60 GSCROLL X,0,8,GH,0

70 GSCROLL X+8,0,8,GH,1

80 NEXT X

100 NEXT I

110 ?TICK()

120 GOTO 120

SAVE 2

#### 7.12 CONFIG システムの設定(システムコマンド)

#### ■ 書式

CONFIG 項目番号, 設定値

#### ■ 引数

項目番号: 0 NTSC の設定

それ以外 無効

設定値 : 選択した項目の設定値(項目により設定値の条件は異なる)

#### ■ 説明

「豊四季タイニーBASIC for Arduino STM32」の環境設定を行います。 現バージョンでは次の設定を行うことができます。

項目	設定名称	設定値	説明
番号			
0	NTSC 垂直同期信号補正	0,1,2	NTSC の垂直同期信号の補正値を設定します。 デフォル
			トは0です。利用するモニターの表示がスクロールして
			今う場合は、映像が乱れる場合は、この値を設定して調
			整してください。
1	キーボード設定	0,1	利用するキーボードのレイアウト設定を行います。
			0:日本語キーボード(デフォルト)
			1: US キーボード
2	自動起動プログラム設定	-1	電源 ON 後に自動起動するプログラムを指定します。
		0~9	0~9 はフラッシュメモリに保存しているプログラム番
			号を指定します。
			-1の設定で自動起動無しの設定となります。

設定はコマンド実行後、直ちに反映されます。本設定は電源を落とすまで有効です。 次回以降の起動にも適用したい場合は、SAVECONFIG コマンドで保存して下さい。

(注意) CONFIG コマンドはプログラム内での実行は出来ません。実行しても無視されます。 直接コマンド入力を行った場合のみ有効です。

### ■ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Illegal value: 指定した項目番号、設定値が正しくない。Overflow: 指定した数値が-32767 ~ 32767 を超えている

## ■ 利用例

利用しているキーボードを US キーボードに設定する。

CONFIG 1,1

OK

SAVECONFIG

#### 7.13 SAVECONFIG システムの設定の保存(システムコマンド)

#### ■ 書式

SAVECONFIG

#### ■ 引数

なし

#### ■ 説明

CONFIG コマンドの設定変更をフラッシュメモリに保存します。次回以降の起動でも設定を有効にします。

フラッシュメモリの利用状況により、保存に失敗する場合があります。その場合は、フラッシュメモリの初期化を EEPFORMAT にて行って下さい。

(注意) SAVECONFIG コマンドはプログラム内での実行は出来ません。実行しても無視されます。 直接コマンド入力を行った場合のみ有効です。

### ■ エラーメッセージ

Syntax error : 書式と異なる利用を行った、プログラム番号に変数、式を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

利用しているキーボードを US キーボードに設定する。

CONFIG 1,1

OK

SAVECONFIG

#### 7.14 REM コメント (一般コマンド)

#### ■ 書式

REM コメント 'コメント

## ■ 引数

コメント:任意の文字列

#### ■ 説明

プログラムに説明等の記載を行います。

(シングルクオート)は REM の省略形です。

REM および、以降の文字以降はすべてコメントとみなし、プログラムとして実行されません。 プログラムの先頭行にコメントを付けた場合は、FILES コマンドで保存プログラム一覧を表示した たきに、各プログラムの見出しとなります。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

### ■ 利用例

LIST

1 REM Print starts

10 I=0:'initialize value

20 PRINT "\*";

30 I=I+1:IF I<10 GOTO 20

40 PRINT

50 END

### 7.15 LET 変数に値を代入(一般コマンド)

#### ■ 書式

LET 変数=値

LET 配列変数=値,値,値,値

変数=値

配列変数=値,値,値,値

### ■ 引数

変数 : 変数 A ~ Z、または配列変数 @(0) ~ @(99)

配列変数 : 配列変数 @(0) ~ @(99)

值 :式、数值、変数、配列変数、数值定数

### ■ 説明

変数に値を代入します。

値は式、数値、定数、変数、配列変数等を評価した整数値です。

配列変数への代入は、複数の値を指定できます。指定した添え字を起点に順番に代入します。

LET @(3)=1,2,3,4,5

上記の記述は、

LET @(3)=1: LET @(4)=2: LET @(5)=3: LET @(5)=4: LET @(5)=5 と同じです。

LET は省略可能です。次の2は同じ結果となります。

LET A=A+1

A=A+1

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

LIST

1 REM Print starts

10 LET I=0:'initialize value

20 PRINT "\*";

30 LET I=I+1:IF I<10 GOTO 20

**40 PRINT** 

50 END

#### 7.16 FREE プログラム領域の残りバイト数の取得(数値関数)

### ■ 書式

FREE()

#### ■ 引数

なし

### ■ 戻り値

プログラム領域の残りバイト数 0 ~ 4095

## ■ 説明

プログラム領域の残りバイト数を返します。

プログラム領域のサイズは 4095 バイトあります。そのうち 1 バイトは終端チェック用に利用しています。残り 4095 バイトがプログラミングに利用出来る容量です。

プログラムサイズは次のように求めることが出来ます。

PRINT 4095-FREE()

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

### ■ 利用例

プログラムサイズを調べる PRINT 4095-FREE() 287 OK

#### 7.17 PRINT 画面への文字表示(一般コマンド)

#### ■ 書式

**PRINT** 

PRINT 文字列|数值

PRINT 文字列|数值;

PRINT 文字列|数值;文字列|数值;···

PRINT 文字列|数值;文字列|数值;…;

PRINT #桁数,文字列 | 数值; 文字列 | 数值;…;

※ | はいずれのうち1つを示す

※ …は可変指定を示す

※;は連結指定、カンマ','も可能

### ■ 引数

文字列: 文字列定数または文字列関数

数値 : 数値定数、変数、配列変数、または数値関数、式

連結指定 : セミコロン';' または カンマ';'

文末に付けると改行抑制される

桁数: #数値 または #-数値 の形式で指定する

例:#3 、 #-3

#### ■ 説明

指定した文字列、数値をカーソル位置に表示します。表示後はカーソルが移動します。 表示要素である文字列、数値は連結指定文字のセミコロン';'、またはカンマ','にて連結して表示する ことが出来ます。PRINT 文の引数の最後に';'または','が付加されている場合は改行しません。

付加されていない場合は引数の内容を表示後、改行します。

複数の数値を連結して表示した場合、デフォルトでは数値と数値の間は空白が入らない詰めた状態で表示します。#数値にて桁数を指定することで、任意の桁数(指定桁に満たない場合は空白を入れる)にて等間隔で表示します。数値の前に・(マイナス)を付加した場合、空白をではなく、0(零)で不足桁を補います。

PRINT 1;":";2;":";3 桁数指定なし

1:2:3

PRINT #2,1;":";2;":";3 桁数指定あり(空白文字で補間)

1: 2: 3

PRINT #-2,1;":";2;":";3 桁数指定あり(0で補間)

01:02:03

# ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行った、引数に数値以外を指定したOverflow: 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

## 時刻を表示する

10 GETTIME A,B,C

20 LOCATE 0,0:PRINT #-2,A;":";B;":";C

30 WAIT 1000

40 GOTO 20

## 実行結果

10:52:00

#### 7.18 INPUT 数値の入力 (一般コマンド)

#### ■ 書式

INPUT 変数

INPUT 変数、オーバーフロー時の既定値

INPUT プロンプト 変数

INPUT プロンプト 変数、オーバーフロー時の既定値

### ■ 引数

変数:入力した値を格納する変数または配列変数

 $A \sim Z, @(0) \sim @(99)$ 

プロンプト : 文字列定数 "プロンプト"

オーバーフロー時の既定値: -32767~32767

#### ■ 説明

現在のカーソル位置にて数値の入力をし、指定した変数にその値を格納します。 キーボードから入力できる文字は数値  $0\sim9$ 、入力訂正の[BS]、[DEL]、入力確定の[Enter]です。 それ以外の入力はできません。

引数に変数のみを指定した場合、"変数名="を表示しその後ろの位置から数値を入力します。 プロンプトを指定した場合は、そのプロンプトを表示しその後ろ位置から数値を入力します。

オーバーフロー時の既定値を指定した場合、入力値した数値がオーバーフローを発生した場合は、オーバーフロー時の既定値を変数に設定します。オーバーフロー時の既定値を設定していない場合は、Overflow エラーとなります。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

INPUT A A=1234 OK

INPUT "Value=" A Value=1234 OK

# 7.19 CLS 画面表示内容の全消去 (一般コマンド)

## ■ 書式

CLS

### ■ 引数

なし

## ■ 説明

画面上に表示している内容を全て消します。

行番号を指定した開始番号から指定した増分で振り直します。

カーソルは画面の先頭左上に移動します。

(注意) 直前に COLOR コマンドにて文字色、背景色の指定を行っている場合、画面全体に背景色が適用されます。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

### ■ 利用例

CLS

### 7.20 CLV 変数領域の初期化(一般コマンド)

### ■ 書式

CLV

#### ■ 引数

なし

### ■ 説明

変数領域(変数、配列)の初期化を行います。

変数 A~Z、配列@(0)~@(99)はすべて 0 に初期化されます。

LRUN コマンドにてプログラムをロードして実行する場合、変数領域は初期化されません。呼び出し元のプログラムの変数をひ引き継ぐことが出来ます。

呼び出されたプログラムにて明示的に変数領域の初期化を行う場合は、CLV コマンドを使って下さい。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

#### ■ 利用例

CLV

#### 7.21 COLOR 文字色の設定(一般コマンド)

#### ■ 書式

COLOR 文字色 COLOR 文字色, 背景色

### ■ 引数

文字色: 色コード 0~9 背景色: 色コード 0~9

#### ■ 説明

文字色の設定を行います。指定した色は以降の文字表示に反映されます。 文字色、背景色で指定する色コードに対する色は次の表の通りです。

表 1 色コード

色コード	色
0	黒
1	赤
2	緑
3	茶
4	青
5	マゼンタ
6	シアン
7	白(デフォルト)
8	黄

(注意) ターミナル版のみ利用可能です。

(注意) 利用するターミナルソフトにより色が正しく表示されない場合があります。 属性指定との併用では正しく表示されない場合があります。 画面を[CTRL-R]、[Page UP]、[Page Down]キーにて再表示した場合、 色情報は欠落します。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 色コードに範囲外の値を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

ランダムな色で\*を表示する

10 FOR I=0 TO 10 20 FOR J=0 TO 10

30 COLOR RND(8): ? "\*";

35 WAIT 100

40 NEXT J

50 ?

60 NEXT I

#### 7.22 ATTR 文字表示属性の設定(一般コマンド)

#### ■ 書式

ATTR 属性

#### ■ 引数

属性: 属性コード 0 ~ 4

#### ■ 説明

文字の表示属性を設定します。指定した表示属性は以降の文字表示に反映されます。 属性に指定する属性コードは次の表の通りです。

表 2 属性コード

属性コード	機能
0	標準(デフォルト)
1	下線
2	反転
3	ブリンク
4	ボールド

(注意) ターミナル版でのみ利用可能です。

(注意) 利用するターミナルソフトにより色が正しく表示されない場合があります。 色指定との併用指定を行った場合、正しく表示されない場合があります。 画面を[CTRL-R]、[Page UP]、[Page Down]キーにて再表示した場合、 色情報は欠落します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 属性コードに範囲外の値を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

Hello,world を反転表示する

10 CLS

20 LOCATE 5,5

30 ATTR 2:? "Hello,world"

40 ATTR 0

#### 7.23 LOCATE カーソルの移動(一般コマンド)

#### ■ 書式

LOCATE 横位置, 縦位置

### ■ 引数

横位置: 画面上の横位置 0 ~ CW-1 縦位置: 画面上の縦位置 0 ~ CH-1

CW、CH は画面の横桁数、縦行数の示す定数です。

利用するフォント等により異なります。

### ■ 説明

カーソルを指定した位置に移動します。

縦横位置それぞれの指定に0以下の数値を指定した場合、それぞれの位置は0となります。 横位置にCW-1を超える数値を指定した場合、横位置はCW-1となります。縦位置にCH-1を超える数値を指定した場合、縦位置はCH-1となります。

定数 CW、CH に設定されている値は次のようすることで確認することができます。

CM?

37

OK

?CH 27

OK

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

画面の指定位置にメッセージを表示する

10 CLS

20 LOCATE 5,5:PRINT "Hello,world."

30 LOCATE 6,6:PRINT "TinyBASIC"

40 LOCATE 7,7:PRINT "Thank you."

#### 実行結果

Hello,world. TinyBASIC Thank you.

### 7.24 VPEEK 画面指定位置の文字コード参照(数値関数)

#### ■ 書式

VPEEK(横位置,縱位置)

### ■ 引数

横位置: 0~CW-1 縦位置: 0~CH-1

CW、CH は画面の横桁数、縦行数の示す定数です。

利用するフォント等により異なります。

## ■ 戻り値

指定位置に表示されている文字の文字コード(0 ~ 255)

### ■ 説明

画面上の指定位置に表示されている文字の文字コードを取得します。 引数の指定位置が範囲外の場合は 0 を返します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

### ■ 利用例

画面最上位の行に表示されている文字を表示する

- 10 FOR I=0 TO CW-1
- 20 C=VPEEK(I,0)
- 30 PRINT CHR\$(C);
- 40 NEXT I
- **50 PRINT**

### 7.25 ABS 絶対値の取得(数値関数)

## ■ 書式

ABS(値)

### ■ 引数

值: -32767 ~ 32767

式、変数、配列変数、数値、数値定数の指定が可能

### ■ 戻り値

指定した値の絶対値

## ■ 説明

指定した値の絶対値を返します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

## ■ 利用例

10 FOR I=-3 TO 3

20 PRINT ABS(I)

30 NEXT I

RUN

3

2

1

0

2

#### 7.26 MAP 数値のスケール変換(数値関数)

#### ■ 書式

MAP(値,範囲下限, 範囲上限, 新範囲下限, 新範囲上限)

#### ■ 引数

値:変換対象の数値-32767 ~ 32767

範囲下限 : 対象数値の数値の下限 -32767 ~ 32767範囲上限 : 対象数値の数値の上限 -32767 ~ 32767

新範囲下限 : 新しい対象数値の数値の下限 -32767 ~ 32767

新範囲上限:新しい新しい対象数値の数値の下限 -32767 ~ 32767

### ■ 戻り値

スケール変換後の値 -32767 ~ 32767

#### ■ 説明

指定した値のスケール変換を行い、その値を返します。

例: A=MAP(ANA(PA07),0,4095,0,99)

上記の例ではアナログ入力値  $0\sim4095$  を  $0\sim99$  のレンジに変換し、その値を返します。 アナログ入力や画面座標指定、PWM パルス出力等にて MAP 関数を使うことで、簡単にレンジ変換を行うことができます。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

### ■ 利用例

## 7.27 RND 乱数の発生(数値関数)

### ■ 書式

RND(値)

#### ■ 引数

值 : 0 ~ 32767

式、変数、配列変数、数値、数値定数の指定が可能

### ■ 戻り値

0から指定した値未満の乱数

## ■ 説明

1から指定した値を含む範囲の乱数を発生させ、その値を返します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

## ■ 利用例

ジャンケンの表示

10 FOR I=1 TO 5

20 R=RND(3)

30 IF R=0 PRINT "Gu"

40 IF R=1 PRINT "Cyoki"

50 IF R=2 PRINT "Paa"

60 NEXT I

RUN

Gu

Paa

Paa

Cyoki

### 7.28 ASC 文字から文字コードへの変換 (数値関数)

#### ■ 書式

ASC(文字列)

ASC(文字列,文字位置)

ASC(変数)

ASC(変数,文字位置)

### ■ 引数

文字列: "文字列"

"ABC"の形式とし、ダブルクォーテーション文字を囲みます

文字位置: 1~32767

変換対象となる左からの文字位置を指定します。

変数: 文字列参照している変数または配列変数

### ■ 戻り値

指定文字に対応する文字コード(0 ~ 255)

### ■ 説明

指定した文字に対応する文字コードを返します。

文字列のみを指定した場合、先頭の文字コードを返します。

10 ?ASC("ABCD")

RUN

65

OK

文字位置を指定した場合、指定した文字のコードを返します。

10 ?ASC("ABCD",3)

RUN

67

OK

変数を指定した場合、変数が参照している文字列のコードを返します。

10 A="ABCDEF"

20 ?ASC(A)

30 ?ASC(A,3)

RUN

65

67

OK

(**注意**) 変数の文字列参照はプログラム中にのみ有効です。コマンドラインでは参照を正しく行う ことが出来ません。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 文字数が不当

'(' or ')' expected : '(' または ')'がない

# ■ 利用例

アルファベット A~Z を表示する

10 C=ASC("A")

20 FOR I=0 TO 25

30 PRINT CHR\$(C+I);

40 NEXT I

50 PRINT

run

ABCDEFGHIJKLMNOPQRSTUVWXYZ

### 7.29 LEN 文字列の長さの取得(数値関数)

### ■ 書式

LEN(変数)

LEN(文字列)

## ■ 引数

変数 : 変数 A~Z または配列変数@(0)~@(99)

文字列: "文字列"の形式(ダブルクォーテーションで囲み)

### ■ 戻り値

文字列の長さ 0 ~ 32767

### ■ 説明

変数で参照した文字列または、直接指定した文字列定数の文字数をカウントし、その値を返します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

### ■ 利用例

### 7.30 CHR\$ 文字コードから文字への変換(文字列関数)

### ■ 書式

CHR\$(文字コード)

#### ■ 引数

文字コード: 0~127,160~255

## ■ 戻り値

指定位置に表示されている文字の文字コード(0 ~ 255)

## ■ 説明

指定した文字コードに対応する文字を返します。 PRINT、SPRINT、GPRINTの引数にて利用可能です。 範囲外の値を指定した場合は空白文字("")を返します。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'が無い

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

アルファベット A~Z を表示する

10 C=ASC("A")

20 FOR I=0 TO 25

30 PRINT CHR\$(C+I);

40 NEXT I

**50 PRINT** 

run

ABCDEFGHIJKLMNOPQRSTUVWXYZ

#### 7.31 STR\$ 変数が参照する文字列の取得・文字列の切り出し

# ■ 書式

STR\$(変数)

STR\$(変数,先頭,長さ)

STR\$(文字列)

STR\$(文字列,先頭,長さ)

### ■ 引数

変数:文字列を参照している変数または配列変数

先頭: 切り出す文字位置先頭

長さ:切り出す文字の長さ

文字列:ダブルクォーテーションで囲った文字列定数("文字列")

# ■ 戻り値

指定した変数が参照している文字列を返します。

指定した文字列の長さを指定した場合はその条件にて文字列を切り出して返します。

# ■ 説明

指定した変数が参照している文字列を全部または切り出して返します。 または指定した文字列定数を全部または切り出して返します。 PRINT、SPRINT、GPRINTの引数にて利用可能です。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'が無い

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

### 7.32 BIN\$ 数値から2進数文字列への変換(文字列関数)

### ■ 書式

BIN\$(数值)

BIN\$(数值,桁数)

# ■ 引数

数値: 変換対象の整数値(-32767 ~ 32767)

桁数: 出力桁数(0~16)

### ■ 戻り値

2 進数文字列(1 桁~16 桁)

### ■ 説明

指定した数値を2進数文字列に変換します。

PRINT、SPRINT、GPRINT の引数にて利用可能です。

析数を指定した場合は数値が指定した桁数に満たない場合は、0で桁を補います。

指定した桁数を超える場合は数値の桁数を優先します。

桁数をしない場合は、先頭の 0 は付加されません。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

Illegal value : 桁数の値が正しくない

Overflow : 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

2 進数で変数の内容を表示する

10 A=1234:B=1

20 PRINT BIN\$(A)

30 PRINT BIN\$(B)

40 PRINT BIN\$(A,4)

50 PRINT BIN\$(B,4)

run

10011010010

1

10011010010

0001

OK

### 7.33 HEX\$ 数値から 16 進数文字列への変換(文字列関数)

# ■ 書式

HEX\$(数值)

HEX\$(数值,桁数)

# ■ 引数

数値: 変換対象の整数値(-32767 ~ 32767)

桁数: 出力桁数 $(0 \sim 4)$ 

# ■ 戻り値

16 進数文字列(1 桁~4 桁)

# ■ 説明

指定した数値を16進数文字列に変換します。

PRINT、SPRINT、GPRINT の引数にて利用可能です。

桁数を指定した場合は数値が指定した桁数に満たない場合は、0 で桁を補います。

指定した桁数を超える場合は数値の桁数を優先します。

桁数をしない場合は、先頭の0は付加されません。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

Illegal value : 桁数の値が正しくない

Overflow : 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

#### 7.34 DMP\$ 整数の小数付き数値文字列への変換(文字列関数)

### ■ 書式

DMP\$(数值

DMP\$(数值,小数点桁数)

DMP\$(数值,小数点桁数,自然部桁数)

### ■ 引数

数值: 変換対象整数 -32767 ~ 32767

小数点析数 : 小数点以下の析数を指定 0 ~ 4(省略時は2)整数部析数 : 小数点以上の析数を指定 0 ~ 8(省略時は0)

# ■ 戻り値

小数点を付加した数値文字列

### ■ 説明

指定した整数を小数点付きの数値に変換します。

PRINT、SPRINT、GPRINT の引数にて利用可能です。

本関数は小数点付き数値の表示を支援します。

引数の小数点以下の桁数を n とした場合、数値 $\div 10^n$ の計算を行い、小数点以下の数値を含めて表示します。

例えば、

PRINT DMP\$(3141,3)

は

3.141

として表示されます。

小数点以下の桁数を指定しない場合、桁数は2として処理します。

整数部桁数を指定した場合、桁数に満たない場合は空白で補完します。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

#### ■ 利用例

アナログ入力値を小数点付き数値で表示する

**10 CLS** 

20 GPIO PA01, ANALOG

30 A=ANA(PA01)

40 LOATE 0,0:PRINT DMP\$(MAP(A,0,4095,0,3300),3)

50 WAIT 200

60 GOTO 30

### 7.35 REDRAW 画面表示の再表示(一般コマンド)

# ■ 書式

REDRAW

### ■ 引数

なし

# ■ 説明

表示用メモリ(VRAM)の内容を画面に再表示を行います。
[CTRL-R]、[Page UP]、[Page Down]キーによる再表示をコマンドにて行います。
再表示においては、カーソルの移動は行いません。

再表示においては、個々に色付しけした文字の色、背景色、属性は欠落します。 再表示後は最後に指定した文字色、背景色、属性に統一されます。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

# ■ 利用例

表示用メモリ(VRAM)に直接書き内容を画面に反映させる 10 POKE VRAM+80\*5+50,ASC("b") 20 REDRAW

### 7.36 INKEY キー入力の読み取り(数値関数)

# ■ 書式

INKEY()

### ■ 引数

なし

# ■ 戻り値

押したキーの文字コード キーが押されていない場合は0

# ■ 説明

キーボード上の押しているキーの文字コードを取得します。 キーが押されていない場合は、0を返します。

(注意) [ESC]、[CTRL-C]はプログラム中断用のため、キー入力の読み取りは出来ません。 また、コンソールの制約により読み取れないキーがあります。

# ■ エラーメッセージ

 Syntax error
 : 文法エラー、書式と異なる利用を行った

 '(' or ')' expected
 : '(' または ')'がない

# ■ 利用例

### 7.37 DATE 現在時刻の表示(一般コマンド)

# ■ 書式

DATE

### ■ 引数

なし

# ■ 説明

内蔵 RTC から現在の時刻を読み、その情報を画面に表示します。

(注意) 内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行って下さい。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

# ■ 利用例

時刻の設定と表示を行う

SETDATE 2017,4,1,12,00,00 OK DATE 2017/04/01 [Sat] 12:00:03 OK

#### 7.38 GETDATE 日付の取得(一般コマンド)

### ■ 書式

GETDATE 年格納変数,月格納変数,日格納変数,曜日格納変数

#### ■ 引数

年 格 納 変 数: 取得した西暦年を格納する変数を指定

月 格 納 変 数: 取得した月を格納する変数を指定 日 格 納 変 数: 取得した日を格納する変数を指定

曜日格納変数: 取得した曜日コードを格納する変数を指定

### ■ 説明

内蔵 RTC から日付情報を取得し、その値を指定した変数に格納します。

格納される値は次の通りです。

年格納変数: 西暦年4桁整数 1900 ~ 2036

月格納変数: 1 ~ 12 日格納変数: 1 ~ 31

曜日格納変数: 曜日コード 0 ~ 6 (0:日 1:月 2:火 3:水 4:木 5:金 6:土)

(注意) 内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行って下さい。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

### ■ 利用例

現在の日付を取得する

DATE
2017/04/01 [Sat] 12:03:05
OK
GETDATE A,B,C,D
OK
PRINT a;b;c;d
2017416
OK

#### 7.39 GETTIME 時刻の取得(一般コマンド)

### ■ 書式

GETTIME 時格納変数,分格納変数 秒格納変数

#### ■ 引数

時格納変数: 取得した時を格納する変数を指定 分格納変数: 取得した分を格納する変数を指定 秒格納変数: 取得した秒を格納する変数を指定

# ■ 説明

内蔵 RTC から日付情報を取得し、その値を指定した変数に格納します。

格納される値を次の通りです。

時格納変数: 0 ~ 23 整数 分格納変数: 0 ~ 59 整数

秒格納変数: 0 ~ 61 整数(うるう秒考慮)

(注意) 内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行って下さい。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

#### ■ 利用例

現在の時間を取得する

```
DATE
2017/04/01 [Sat] 12:10:51
OK
GETTIME A,B,C
OK
PRINT #-2,a;":";b;":";c
12:11:01
OK
```

(補足) PRINT 文の #-2 は数値を 2 桁(0 付き)で表示する指定です。

### 7.40 SETDATE 時刻の設定(一般コマンド)

### ■ 書式

SETDATE 年,月,日,時,分,秒

### ■ 引数

年:1900 ~ 2036 西暦年4桁の整数

月:1~12整数日:1~31整数時:0~23整数分:0~59整数

秒:0~61 整数(うるう秒考慮)

### ■ 説明

指定した時刻を内蔵 RTC に設定します。

(注意) 内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行って下さい。

# ■ エラーメッセージ

Illegal value : 指定した引数の数値が有効範囲以外

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

時刻の設定と表示を行う

Y=2017:M=4:D=1:H=12:N=0:S=0 SETDATE Y,M,D,H,N,S

OK

DATE

2017/04/01 [Sat] 12:00:03

OK

#### 7.41 TICK 起動からの経過時間取得(数値関数)

### ■ 書式

TICK()

 $TICK(\mp - F)$ 

# ■ 引数

モード: LOW (または 0): 下位 15 ビットの値取得

HIGH (または 1): 上位 15 ビットの値取得

### ■ 戻り値

モード LOW時: 0 ~ 32767 (0.1 秒単位) 約 54.6 分迄

モード HIGH時: 0 ~ 32767 (32767 秒単位) 約 54.6 分×取得数値

# ■ 説明

起動からの経過時間を返します。

モード指定無、または LOW 指定の場合は 0.1 秒単位の経過時間を返します。

モード指定が HIGH 場合は上位のビットを返します。

LOW の桁は約 54.6 分で桁溢れして 0 に戻ります。桁溢れの発生を判定する場合は HIGH 指定にて上位ビットを参照して下さい。

秒単であれば、内蔵 RTC を使う GETTIME を使う方法もあります。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

5 秒間時間待ちループを行う

10 RESETTICK

20 IF TICK()<50 GOTO 20

30 PRINT "5sec passed"

# 7.42 RESETTICK 起動からの経過時間カウントのリセット(一般コマンド)

# ■ 書式

RESETTICK

### ■ 引数

なし

# ■ 説明

起動からの経過時間カウンタをリセット(0 を設定)します。 TICK()関数使う前に経過時間カウントをリセットすることで 0 からのカウントを行うことが出来ます。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

# ■ 利用例

5 秒間時間待ちループを行う

10 RESETTICK

20 IF TICK()<50 GOTO 20

30 PRINT "5sec passed"

7.43WAIT 時間待ち (一般コマンド)

### ■ 書式

WAIT 待ち時間(ミリ秒)

### ■ 引数

待ち時間: 0 ~ 32767 (単位 ミリ秒)

### ■ 説明

引数で指定した時間(ミリ秒単位)、時間待ち(ウェイト)を行います。

最大で 32767 ミリ秒 (32.8 秒) の時間待ちが可能です。

長い時間待ちを行う必要がある場合は、TICK()や GETTIME を使った方法を検討してください。

(注意) 時間待ち中はキー操作によるプログラム中断を行うことは出来ません。 短い時間の指定にてループ処理を行う等の対策を行ってください。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 待ち時間に範囲外の値を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

画面上の指定位置に時刻を1秒間隔で更新表示する

10 SETDATE 2017,4,1,12,0,0

20 CLS

30 LOCATE 5,5

40 DATE

50 WAIT 1000

60 GOTO 30

#### 7.44 GPIO GPIO 機能設定 (一般コマンド)

### ■ 書式

GPIO ピン番号、モード

#### ■ 引数

ピン番号:0~34

ピン番号は数値の他に次のピン名(定数)での指定も可能です。 PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,

PA09, PA10, PA11, PA12, PA13, PA14, PA15,

PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,

PB09, PB10, PB11, PB12, PB13, PB14, PB15,

PC13, PC14, PC15

### モード:次の定数

OUTPUT\_OD: デジタル出力(オープンドレイン)

OUTPUT : デジタル出力

INPUT FL : デジタル入力 (フロート状態: Arduino の INPUT 指定と同じ)

INPUT\_PU : デジタル入力 (内部プルアップ抵抗有効) INPUT PD : デジタル (内部プルダウン抵抗有効)

ANALOG: アナログ入力 PWM: PWM 出力

### ■ 説明

ボード上の GPIO ピンの機能設定を行います。

Arduino の pinMode()に相当します。

GPIO ピンを使って信号の入出力を行う場合は、必ず本コマンドによる設定が必要です。

(**注意**) ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。 ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : ピン番号、モードに範囲外の値を指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

Illegal value : 指定した数値が不当である

#### ■ 利用例

PB01 ピンからアナログ入力値を読み取り、その値を画面に随時表示します。

**10 CLS** 

20 GPIO PB01, ANALOG

30 A=ANA(PB01)

40 LOCATE 5,5: ? A; " "

50 GOTO 30

7.45 OUT デジタル出力 (一般コマンド)

### ■ 書式

OUT ピン番号、出力値

#### ■ 引数

ピン番号:0~34

ピン番号は数値の他に次のピン名(定数)での指定も可能です。 PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08, PA09, PA10, PA11, PA12, PA13, PA14, PA15, PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,

PB09, PB10, PB11, PB12, PB13, PB14, PB15,

PC13, PC14, PC15

### 出力值:

LOW または 0 : 0V を出力する HIGH or 0 以外の値: 3.3V を出力する

### ■ 説明

指定ピンから、指定した出力を行います。

出力を行う場合は事前に GPIO コマンドによる機能設定(出力設定)が必要です。

(注意) ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。 ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

### ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: ピン番号、モードに範囲外の値を指定したOverflow: 指定した数値が-32767 ~ 32767 を超えている

### ■ 利用例

Blue Pill ボードの搭載 LED (PC13 ピン) を点滅させます。

10 P=PC13

20 GPIO P,OUTPUT

30 OUT P, HIGH

40 FOR I=1 TO 10

50 OUT P,LOW

60 WAIT 300

70 OUT P, HIGH

80 WAIT 300

90 NEXT I

#### 7.46 POUT PWM パルス出力 (一般コマンド)

### ■ 書式

POUT ピン番号, デューティー値 POUT ピン番号, デューティー値, 周波数

# ■ 引数

ピン番号: 下記のピンが利用可能です。

(後で記入)

ピン番号は数値の他に次のピン名(定数)での指定も可能です。

デューティー値: 0 ~ 4095

0 がデューティー比 100%となります。4095 がデューティー比 100%となります。

周波数: 0 ~ 32767 (単位 Hz)

指定しない場合は 490Hz となります(Arduino Uno 互換)

### ■ 説明

指定ピンから、PWM パルス出力を行います。

出力を行う場合は事前に GPIO コマンドによる PWM 設定(出力設定)が必要です。 パルス出力を停止する場合は、デューティー値に 0 を指定して下さい。

**(注意) ピン番号の指定範囲及び定数**は、ST32F103C8T6 での利用を想定したものです。 ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

### ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: ピン番号、モードに範囲外の値を指定したOverflow: 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

アナログジョイスティックでサーボーモーターを制御を行う

- 1 'サーホ゛モーターセイキ゛ョ
- 5 CLS
- 10 GPIO PB00, ANALOG
- 20 GPIO PA08,PWM
- 30 P=MAP(ANA(PB00),0,4095,102,491)
- 40 POUT PA08, P, 50
- 45 D=MAP(P,102,491,-90,90)
- 47 LOCATE 0,0:?#3,D
- 50 GOTO 30

### 7.47 IN デジタル入力 (数値関数)

### ■ 書式

IN(ピン番号)

#### ■ 引数

ピン番号:0~34

ピン番号は数値の他に次のピン名(定数)での指定も可能です。 PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08, PA09, PA10, PA11, PA12, PA13, PA14, PA15, PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08, PB09, PB10, PB11, PB12, PB13, PB14, PB15, PC13, PC14, PC15

# ■ 戻り値

取得した値 0(LOW) または 1(HIGH)

# ■ 説明

指定ピンの入力値を読み取り、その値を返します。 入力を行う場合は事前に GPIO コマンドによる機能設定(入力設定)が必要です。

**(注意) ピン番号の指定範囲及び定数**は、ST32F103C8T6 での利用を想定したものです。 ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

# ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: ピン番号、モードに範囲外の値を指定したOverflow: 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'がない

### ■ 利用例

#### 7.48 ANA アナログ入力 (数値関数)

### ■ 書式

ANA(ピン番号)

### ■ 引数

ピン番号: 0 ~ 7,16,17 または以下の定数 PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PB00, PB0

### ■ 戻り値

取得した値 0~4095(12 ビット)

### ■ 説明

指定ピンのアナログ入力値を読み取り、その値を返します。 アナログ入力を行う場合は事前に GPIO コマンドによる機能設定 (アナログ入力) が必要です。

**(注意) ピン番号の指定範囲及び定数**は、ST32F103C8T6 での利用を想定したものです。 ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

# ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: ピン番号、モードに範囲外の値を指定したOverflow: 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'がない

#### ■ 利用例

PB01 ピンからアナログ入力値を読み取り、その値を画面に随時表示します。

10 CLS

20 GPIO PB01, ANALOG

30 A=ANA(PB01)

40 LOCATE 5,5: ? A; " "

50 GOTO 30

#### 7.49 SHIFTIN デジタルシフトアウト入力 (数値関数)

#### ■ 書式

SHIFTIN(データピン番号、クロックピン番号、入力形式) SHIFTIN(データピン番号、クロックピン番号、入力形式、条件)

# ■ 引数

データピン番号 : 0 ~ 34 データを入力するピン

クロックピン番号 : 0 ~ 34 クロックを出力するピン

上記のピン番号は数値の他に次のピン名(定数)での指定も可能です。

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,

PA09, PA10, PA11, PA12, PA13, PA14, PA15,

PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,

PB09, PB10, PB11, PB12, PB13, PB14, PB15,

PC13, PC14, PC15

入力形式: 入力するデータの順番を下記にて指定

LSB または 0 下位ビットから入力する MSB または 1 上位ビットから入力する

条件: LOW または HIGH

データピンからのデータを読み取るクロックのタイミングを指定します。

 LOW:
 クロックが LOW の場合にデータを読み取る。

 HIGH:
 クロックが HIGH の場合にデータを読み取る。

#### ■ 戻り値

入力値(1バイト)

#### ■ 説明

クロックにて同期を行い、データピンから1バイト分のデータを1ビットずつ入力します。 Arduino の shiftIn()と同等の動作をします。

引数に条件を指定した場合、クロックが指定した状態の時にデータピンからデータを読み取ります。 条件を指定していない場合は、条件は HIGH(Arduino の shiftIn()と同様の仕様)となります。

データピン、クロックピンは事前に GPIO コマンドによる機能設定 (デジタル入出力) が必要です。

(注意) ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。

ST32F103C8T6以外のMPUにて使う場合は、機能設定できない場合があります。

### ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: ピン番号、モードに範囲外の値を指定した

# 豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'がない

# ■ 利用例

ファミコン用ゲームパッドからボタン操作情報を取得する

- 10 CLS
- 20 GPIO PB12, INPUT\_FL
- 30 GPIO PB13,OUTPUT
- 40 GPIO PB14, OUTPUT
- 50 OUT PB13, HIGH
- 60 OUT PB13, LOW
- 70 R=SHIFTIN(PB12,PB14,LSB,LOW)
- 80 LOCATE 0,0:?BIN\$(R,8)
- 90 WAIT 100
- 100 GOTO 50

### 7.50 SHIFTOUT デジタルシフトアウト出力(一般コマンド)

#### ■ 書式

SHIFTOUT データピン番号、クロックピン番号、出力形式、出力データ

# ■ 引数

データピン番号:  $0 \sim 34$  データを出力するピン

クロックピン番号: 0~34 クロックを出力するピン

上記のピン番号は数値の他に次のピン名(定数)での指定も可能です。

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,

PA09, PA10, PA11, PA12, PA13, PA14, PA15,

PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,

PB09, PB10, PB11, PB12, PB13, PB14, PB15,

PC13, PC14, PC15

出 カ 形 式:出力するデータの順番を下記にて指定

LSB または 0: 下位ビットから出力する MSB または 1: 上位ビットから出力する

出 カ デ ー タ: 出力するデータ(下位8ビットのみ有効)

### ■ 説明

クロックにて同期を行い、データピンから1バイト分のデータを1ビットずつ出力します。 Arduino の shiftOut()と同等の動作をします。

データピン、クロックピンは事前に GPIO コマンドによる機能設定(デジタル出力)が必要です。

(注意) ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。 ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

#### ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: ピン番号、モードに範囲外の値を指定したOverflow: 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

シフトレジスタ 74HC595 を使って LED8 個を制御する

10 GPIO PB12,OUTPUT

20 GPIO PB13,OUTPUT

30 GPIO PB14, OUTPUT

40 D=\$80

50 FOR I=0 TO 6

60 GOSUB "led"

- 70 D=D>>1
- 80 WAIT 200
- 90 NEXT I
- 100 FOR I=0 TO 6
- 110 GOSUB "led"
- 120 D=D<<1
- 130 WAIT 200
- 140 NEXT I
- 150 GOTO 40
- 160 "led"
- 170 OUT PB13,LOW
- 180 SHIFTOUT PB12, PB14, MSB, D
- 190 OUT PB13,HIGH
- 200 TONE 600,10
- 210 RETURN

7.51 SMODE シリアルポート機能切り替え (一般コマンド)

#### ■ 書式

SMODE モード指定

SMODE モード指定, "通信速度"

# ■ 引数

モード指定:0(デフォルト) USB ポート= 対話利用、 シリアルポート= データ通信

1 USB ポート= データ通信、シリアルポート= 対話利用

2 シリアル利用禁止

通信速度 : 文字列形式の数値 "110"~ "921600" ※ モード指定が1の場合のシリアルポートの

通信速度の指定を行う。

### ■ 説明

USB シリアルと GPIO ピンのシリアルポートの機能を入れ替えます。

デフォルトのモード指定 0 では、USB ポートは対話利用(入力、プログラム転送、デバッグ)、GPIO シリアルポート (PA9、PA10) はデータ通信用の設定です。

モード指定 1 では、GPIO シリアルポートをオープンし、対話利用可能な状態となります。USB ポートはデータ通信ためのコマンドが利用可能になります。ただしシリアル通信は常にオープンされている状態となります。SOPEN、SCLOSE の実行は無視されます。通信速度の設定は行うことができません。

モード指定2では、シリアル利用を禁止します。

GPIO シリアルポートはモードを 1 に切り替えた時点でシリアル通信はオープンされます。モード 0 に切り替えた場合、シリアル通信はクローズされます。

シリアル通信をデータ通信に設定した場合は次のコマンドを利用することができます。

SOPEN : シリアル通信開始

SERITE : 1 バイト送信 SREAD : 1 バイト受信

SREADY : 受信データ有無の確認

SPRINT : 文字列出力(PRINT と同等の出力)

SCLOSE : シリアル通信クローズ

(注意) シリアル通信をオープンした状態で GPIO コマンドにて PA9,PA10 に I/O 機能の割り当てを行った場合、正しい動作を行うことが出来ません。必ず SCLOSE にてシリアル通信をクローズしてからピンに機能を割り当て下さい。

# ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: 指定した引数に範囲外の値を指定した

豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

Overflow : 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

SMODE 1

### 7.52 SOPEN シリアル通信オープン (一般コマンド)

### ■ 書式

SOPEN 通信速度

#### ■ 引数

通信速度 : 文字列形式の数値 "110"~ "921600"

例: "115200"

### ■ 説明

データ通信設定したシリアルポートのシリアル通信の利用を開始します。

通信速度は文字列形式の数値で指定します。

シリアル通信は次の GPIO ピンまたは、USB ポートを利用することが出来ます。

PA9 : TX 送信 PA10 : RX 受信

シリアルポートが既にオープン状態の場合は、シリアルポートをクローズしてからオープンします。 シリアル通信の利用には次のコマンドが用意されています。

SMODE : GPIO シリアルポート、USB シリアルポート機能切り替え

SOPEN : シリアル通信開始

SERITE : 1 バイト送信 SREAD : 1 バイト受信

SREADY : 受信データ有無の確認

SPRINT : 文字列出力(PRINT と同等の出力)

SCLOSE : シリアル通信クローズ

(注意) シリアル通信をオープンした状態で GPIO コマンドにて PA9,PA10 に I/O 機能の割り当てを行った場合、正しい動作を行うことが出来ません。必ず SCLOSE にてシリアル通信をクローズしてからピンに機能を割り当て下さい。

# ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったIllegal value: 指定した引数に範囲外の値を指定した

#### ■ 利用例

シリアル通信経由でテキストメッセージを送信する

10 SOPEN "9600"

20 SPRINT "Hello, TinyBASIC"

30 SCLOSE

### 7.53 SCLOSE シリアル通信クローズ (一般コマンド)

### ■ 書式

SCLOSE 通信速度

### ■ 引数

なし

### ■ 説明

データ通信設定したシリアルポートのシリアル通信を終了します。

シリアル通信は次の GPIO ピンまたは、USB ポートを利用することが出来ます。

PA9 : TX 送信 PA10 : RX 受信

シリアル通信の利用には次のコマンドが用意されています。

SOPEN: シリアル通信開始SERITE: 1 バイト送信SREAD: 1 バイト受信

SREADY : 受信データ有無の確認

SPRINT : 文字列出力(PRINT と同等の出力)

SCLOSE : シリアル通信クローズ

(注意) シリアル通信をオープンした状態で GPIO コマンドにて PA9,PA10 に I/O 機能の割り当てを行った場合、正しい動作を行うことが出来ません。必ず SCLOSE にてシリアル通信をクローズしてからピンに機能を割り当て下さい。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

#### ■ 利用例

シリアル通信経由でテキストメッセージを送信する

10 SOPEN "9600"

20 SPRINT "Hello, TinyBASIC"

30 SCLOSE

#### 7.54 SPRINT シリアル通信文字列出力(一般コマンド)

### ■ 書式

**PRINT** 

PRINT 文字列|数值

PRINT 文字列 | 数值;

PRINT 文字列|数值;文字列|数值;···

PRINT 文字列|数值;文字列|数值;…;

PRINT #桁数,文字列 | 数值; 文字列 | 数值;…;

※ | はいずれのうち1つを示す

※ …は可変指定を示す

※;は連結指定、カンマ','も可能

# ■ 引数

文字列: 文字列定数または文字列関数

数値 : 数値定数、変数、配列変数、または数値関数、式

連結指定 : セミコロン';' または カンマ';'

文末に付けると改行抑制される

桁数:#数値 または #-数値 の形式で指定する

例:#3 、 #-3

### ■ 説明

指定した文字列、数値をデータ通信設定したシリアルポートに出力します。 書式は PRINT 文と同じです。

表示要素である文字列、数値は区切り文字のセミコロン';'、またはカンマ','にて連結して表示することが出来ます。

連結表示において、数値は桁指定により指定した桁幅にて等間隔で表示します。

析指定においてマイナスの数値を指定した場合は、間隔を 0 で補完します。

正の数値の場合は空白文字で補完します。

PRINT 文の引数の最後に','または','が付加されている場合は改行しません。 付加されていない場合は引数の内容を表示後、改行します。

シリアル通信は次の GPIO ピンを利用します。

PA9 : TX 送信 PA10 : RX 受信

シリアル通信の利用には次のコマンドが用意されています。

### 豊四季タイニーBASIC for Arduino STM32 V0.82 リファレンスマニュアル RV0.1

SOPEN :シリアル通信開始 SERITE : 1 バイト送信 SREAD : 1 バイト受信

SREADY : 受信データ有無の確認

SPRINT : 文字列出力(PRINT と同等の出力)

SCLOSE :シリアル通信クローズ

(注意) シリアル通信をオープンした状態で GPIO コマンドにて PA9,PA10 に I/O 機能の割り当て を行った場合、正しい動作を行うことが出来ません。必ず SCLOSE にてシリアル通信をク ローズしてからピンに機能を割り当て下さい。

# ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったOverflow: 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

シリアル通信経由でテキストメッセージを送信する

10 SOPEN "9600"

20 SPRINT "Hello, TinyBASIC"

30 SCLOSE

#### 7.55 SWRITE シリアル通信 1 バイト出力 (一般コマンド)

### ■ 書式

SERITE 送信データ

### ■ 引数

送信データ : 0~255 (1 バイトデータ、変数指定時は上位 8 ビットは無視)

### ■ 説明

データ通信設定したシリアルポートに1バイト分のデータを出力します。

シリアル通信は次の GPIO ピンを利用します。

PA9 : TX 送信 PA10 : RX 受信

シリアル通信の利用には次のコマンドが用意されています。

 SOPEN
 : シリアル通信開始

 SERITE
 : 1 バイト送信

 SREAD
 : 1 バイト受信

SREADY : 受信データ有無の確認

SPRINT : 文字列出力(PRINT と同等の出力)

SCLOSE : シリアル通信クローズ

(注意) シリアル通信をオープンした状態で GPIO コマンドにて PA9,PA10 に I/O 機能の割り当てを行った場合、正しい動作を行うことが出来ません。必ず SCLOSE にてシリアル通信をクローズしてからピンに機能を割り当て下さい。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

シリアルポートに'A'から'Z'の文字コードを出力する

10 SOPEN "115200"

20 FOR C=ASC("A") TO ASC("Z")

30 SWRITE C

40 NEXT C

50 SCLOSE

### 7.56 SREADY シリアル通信データ受信データ確認 (数値関数)

### ■ 書式

SREADY()

#### ■ 引数

なし

### ■ 戻り値

0: 受信データなし 1: 受信データあり

#### ■ 説明

シリアルポートに受信可能なデータが着信しているかを確認します。

シリアルポートから SREAD 関数にてデータの受信を行う際に、事前にデータの有無を確認することが出来ます。

シリアル通信は次の GPIO ピンを利用します。

PA9 : TX 送信 PA10 : RX 受信

シリアル通信の利用には次のコマンドが用意されています。

SOPEN: シリアル通信開始SERITE: 1 バイト送信SREAD: 1 バイト受信

SREADY: 受信データ有無の確認

SPRINT : 文字列出力(PRINT と同等の出力)

SCLOSE : シリアル通信クローズ

(注意) シリアル通信をオープンした状態で GPIO コマンドにて PA9,PA10 に I/O 機能の割り当てを行った場合、正しい動作を行うことが出来ません。必ず SCLOSE にてシリアル通信をクローズしてからピンに機能を割り当て下さい。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

# ■ 利用例

受信した文字列を画面に表示する

(".":終了、\$OD:改行、それ以外の制御文字は無視)

10 SOPEN "115200"
20 IF SREADY()=0 GOTO 20
30 D=SREAD()
40 IF D=ASC(".") GOTO 80
50 IF D=\$0D ?
60 IF D>=32 ? CHR\$(D);
70 GOTO 20
80 SCLOSE
90 ? :? "Done."

#### 7.57 SREAD シリアル通信 1 バイト受信 (数値関数)

### ■ 書式

SREAD()

#### ■ 引数

なし

### ■ 戻り値

受信データ : 0 ~ 255、データなしの場合は-1

# ■ 説明

データ通信設定したシリアルポートから1バイトデータを受信します。

データが無い場合は-1を返します。データが無い場合の受信待ちは行いません。

シリアル通信は次の GPIO ピンを利用します。

PA9 : TX 送信 PA10 : RX 受信

シリアル通信の利用には次のコマンドが用意されています。

SOPEN: シリアル通信開始SERITE: 1 バイト送信SREAD: 1 バイト受信

SREADY : 受信データ有無の確認

SPRINT : 文字列出力(PRINT と同等の出力)

SCLOSE : シリアル通信クローズ

(注意) シリアル通信をオープンした状態で GPIO コマンドにて PA9,PA10 に I/O 機能の割り当てを行った場合、正しい動作を行うことが出来ません。必ず SCLOSE にてシリアル通信をクローズしてからピンに機能を割り当て下さい。

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

### ■ 利用例

受信した文字列を画面に表示する(".":終了、\$OD:改行、それ以外の制御文字は無視)

10 SOPEN "115200"

20 IF SREADY()=0 GOTO 20

30 D=SREAD()

40 IF D=ASC(".") GOTO 80

50 IF D=\$0D`?

60 IF D>=32 ? CHR\$(D);

70 GOTO 20

80 SCLOSE

90 ? :? "Done."

7.58 I2CR I2C スレーブデバイスからのデータ受信(数値関数)

#### ■ 書式

I2CR(デバイスアドレス、コマンドアドレス、コマンド長、データアドレス、データ長)

#### ■ 引数

デバイスアドレス : 0 ~ 127(\$00 ~ \$7F)(7 ビット指定)

I2C スレーブアドレスを 7 ビット形式で指定

コマンドアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)

送信するコマンドが格納されている SRAM 内相対アドレス

コマンド長 : 0 ~ 32767

送信するコマンドのバイト数

受信データアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)

受信データを格納する SRAM 内相対アドレス

データ長 : 0 ~ 32767

送信するデータのバイト数

#### ■ 説明

I2C スレーブデバイスからデータを受信します。

送信先はデバイスアドレスにて I2C スレーブアドレスを指定します。

受信において、コマンド等の制御データの送信が必要な場合は、コマンドアドレス、コマンド長にて送信するデータを指定します。受信のみを行う場合は、コマンドアドレス、コマンド長に 0 を設定します。送信受信するデータは SRAM 先頭からの相対アドレスにて指定します。

相対アドレスの指定には次の定数を利用することで有用な領域への書込み・参照が簡単に行えます。

MEM : ユーザーワーク領域 サイズ 2048 バイト

VRAM : 画面表示用メモリ (CW×CH) サイズ利用フォントにより可変

VAR : 変数領域 (A~Z) サイズ 52 バイト ARRAY : 配列変数領域(@(0)~@(99)) サイズ 200 バイト

I2C 通信には次の接続ピンを利用します。

PB6 : SCL (I2C クロック) PB7 : SDA (I2C データ)

**(注意)** SRAM 容量を超える相対アドレス (STM32F103C8T6 の場合は 20480) への参照はエラーとなります。

#### ■ 戻り値

0:正常終了

1:通信バッファに対してデータが長すぎる

2: アドレス送信に NACK が返された

3: データ送信に NACK が返された

# 4:その他のエラー

# ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 指定した引数の値が不当である

Out of range value : 指定した値が有効範囲を超えている

Overflow : 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

```
I2C EEPROM(AT24C256 スレーブアドレス $50) にデータの読み書きを行う
```

```
100 POKE MEM+0,$00,$00
110 POKE MEM+2,64,65,66,67
120 R=I2CW($50,MEM,2,MEM+2,4)
130 ? "snd r=";R
140 POKE MEM+6,0,0,0
150 WAIT 5
160 R=I2CR($50,MEM,2,MEM+6,4)
170 ? "rcv r=";R
180 ? "rcv data:";
190 FOR I=0 TO 3
200 ? PEEK(MEM+6+I);" ";
210 NEXT I
220 ?
```

### 実行結果

run snd r=0 rcv r=0 rcv data:64 65 66 67 OK 7.59 I2CW I2C スレーブデバイスへのデータ送信(数値関数)

### ■ 書式

I2CW(デバイスアドレス、コマンドアドレス、コマンド長、データアドレス、データ長)

#### ■ 引数

デバイスアドレス: 0 ~ 127(\$00 ~ \$7F)(7 ビット指定)

I2C スレーブアドレスを 7 ビット形式で指定

コマンドアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)

送信するコマンドが格納されている SRAM 内相対アドレス

コマンド長 : 0 ~ 32767

送信するコマンドのバイト数

データアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)

送信するデータが格納されている SRAM 内相対アドレス

データ長 : 0 ~ 32767

送信するデータのバイト数

#### ■ 説明

I2C スレーブデバイスにデータを送信します。

送信先はデバイスアドレスにて I2C スレーブアドレスを指定します。

送信するデータは SRAM 先頭からの相対アドレスにて指定します。

送信するデータはあらかじめ、設定しておく必要があります。

コマンドとデータの区別はありません。デバイスに対しては、単純にコマンド、データの順に送信しています。

相対アドレスの指定には次の定数を利用することで有用な領域への書込み・参照が簡単に行えます。

MEM : ユーザーワーク領域 サイズ 2048 バイト

VRAM : 画面表示用メモリ (CW×CH) サイズ利用フォントにより可変

VAR : 変数領域 (A~Z) サイズ 52 バイト ARRAY : 配列変数領域(@(0)~@(99)) サイズ 200 バイト

I2C 通信には次の接続ピンを利用します。

PB6 : SCL (I2C クロック) PB7 : SDA (I2C データ)

**(注意)** SRAM 容量を超える相対アドレス (STM32F103C8T6 の場合は 20480) への参照はエラーとなります。

### ■ 戻り値

0:正常終了

1:通信バッファに対してデータが長すぎる

2: アドレス送信に NACK が返された

3: データ送信に NACK が返された

4:その他のエラー

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 指定した引数の値が不当である

Out of range value : 指定した値が有効範囲を超えている

Overflow : 指定した数値が-32767 ~ 32767 を超えている

# ■ 利用例

```
I2C EEPROM(AT24C256 スレーブアドレス $50) にデータの読み書きを行う
 100 POKE MEM+0,$00,$00
 110 POKE MEM+2,64,65,66,67
 120 R=I2CW($50,MEM,2,MEM+2,4)
 130 ? "snd r=";R
 140 POKE MEM+6,0,0,0,0
 150 WAIT 5
 160 R=I2CR($50,MEM,2,MEM+6,4)
 170 ? "rcv r=";R
 180 ? "rcv data:";
 190 FOR I=0 TO 3
 200 ? PEEK(MEM+6+I);" ";
 210 NEXT I
 220 ?
 実行結果
 run
 snd r=0
```

# 接続している I2C スレーブを調べる

rcv data:64 65 66 67

rcv r=0

OK

done. OK

(補足) I2C スレーブアドレスのみ送信し、正常終了(ACK を返した)のアドレスを調べています。

```
10 FOR I=0 TO $7F
20 C=I2CW(I,MEM,0,MEM,0)
30 IF C=0 PRINT HEX$(I,2);" ";
40 NEXT I
50 PRINT :PRINT "done."

実行結果
run
50
```

7.60 PEEK 指定アドレスの値参照 (数値関数)

### ■ 書式

PEEK(相対アドレス)

#### ■ 引数

相対アドレス: 参照を行う SRAM 領域先頭からの相対アドレス(16 ビット)

#### ■ 戻り値

指定した相対アドレス内の1バイトデータ(0~255)

## ■ 説明

SRAM 先頭からの相対アドレスを指定し、格納されている値(1バイト)を返します。 相対アドレスの指定には次の定数を利用することで有用な領域への参照が簡単に行えます。

MEM : ユーザーワーク領域 サイズ 2048 バイト

VRAM : 画面表示用メモリ (CW×CH) サイズ利用フォントにより可変

VAR : 変数領域 (A~Z) サイズ 52 バイト ARRAY : 配列変数領域(@(0)~@(99)) サイズ 200 バイト

ユーザーワーク領域は利用者が自由に利用出来る領域です。

それ以外の領域は BASIC の実行にて利用する領域です。

例として画面上のカーソル位置 X,Y に格納されている文字の参照は次のようになります。

C = PEEK(VRAM+Y\*80+X)

(注意) SRAM の領域外のアドレスを指定した場合、エラーとなりなす。

領域チェックにはマクロ SRAM\_SIZE の定義値(デフォルト 20480) を利用しています。 デフォルト値よりも大きい容量のマイコンを利用している場合はこの定義値を修正して下 さい。

## エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったOut of range value: SRAM の領域外のアドレスを指定した

'(' or ')' expected : 括弧の指定が正しくない

Overflow : 指定した数値が-32767 ~ 32767 を超えている

## ■ 利用例

変数AからZをユーザーワーク領域に保存する例

10 FOR I=0 TO 51

20 POKE MEM+I, PEEK(VAR+I)

30 NEXT I

7.61 POKE 指定アドレスへのデータ書き込み (一般コマンド)

#### ■ 書式

POKE 相対アドレス,データ

POKE 相対アドレス,データ,データ, ... データ (可変個数指定)

#### ■ 引数

相対アドレス: 参照を行う SRAM 領域先頭からの相対アドレス(16 ビット)

データ: 書き込むデータ(下位8じットのみ有効)

## ■ 説明

SRAM 先頭からの指定した相対アドレスに指定したデータを書き込みます。

相対アドレスの指定には次の定数を利用することで有用な領域への書込みが簡単に行えます。

MEM : ユーザーワーク領域 サイズ 2048 バイト

VRAM : 画面表示用メモリ (CW×CH) サイズ利用フォントにより可変

VAR: 変数領域 (A~Z)サイズ 52 バイトARRAY: 配列変数領域(@(0)~@(99) )サイズ 200 バイト

ユーザーワーク領域は利用者が自由に利用出来る領域です。

それ以外の領域は BASIC の実行にて利用する領域です。

指定した相対アドレスに上記以外の領域を指定した場合はエラーとなります。

例として画面上のカーソル位置 X,Y への文字の書込みは次のようになります。 POKE VRAM+Y\*80+X, ASC("A")

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Out of range value : SRAM の領域外のアドレスを指定した

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

変数AからZをユーザーワーク領域に保存する例

10 FOR I=0 TO 51

20 POKE MEM+I, PEEK(VAR+I)

30 NEXT I

7.62 EEPFORMAT 仮想 EEPROM のフォーマット(一般コマンド)

#### ■ 書式

**EEPFORMAT** 

#### ■ 引数

なし

#### ■ 説明

仮想 EEPROM のフォーマットを行います。

仮想 EEPROM とは内部フラッシュメモリの一部をソフトウェアにて EEPROM のように利用しますものです。仮想 EEPROM に書き込んだデータは、マイコンボードの電源を落としても保持されます。

EEPFORMAT コマンドは仮想的に EEPROM として利用するためにフラッシュメモリの初期化処理を行います。既に EEPWTIRE コマンドにてデータの書込みを行っている場合、そのデータを含めて初期化されます。

仮想 EEPROM はフラッシュメモリの2ページ(1024 バイト×2)を利用しています。

フラッシュメモリの書込み回数の制約を考慮し、同じアドレスへの繰り返し書き込みに対して書き 込み位置を分散しています。

仮想 EEPROM へのデータ保存は2バイト単位(変数のバイト数と同じ)です。

保存可能なデータ量は 250 ワード(250 バイト)となります。

EEPWRITE、EEPREAD コマンドで指定する読書きを指定するアドレスは実際にはアドレスではなく、検索キーです。0~32767の間に任意で利用出来ます。

検索キー数はデータ保存量が250ワードの制約から、250以内に抑える必要あります。仮想EEPROMの機構から数十以内に抑えることを推奨します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

## ■ 利用例

EEPFORMAT OK 7.63 EEPWRITE 仮想 EEPROM のへのデータ書き込み(一般コマンド)

#### ■ 書式

EEPWRITE アドレス,データ

#### ■ 引数

アドレス : 仮想 EEPROM 内アドレス 0 ~ 32767

データ: 書き込みデータ 32767 ~ 32767 (2バイト分)

\$0000 ~ \$FFFF

#### ■ 説明

仮想 EEPROM の指定したアドレスにデータを書き込みます。

書き込み単位は 2 バイトです。仮想 EEPROM に初めてデータ書き込みを行う場合は事前に EEPFORMAT コマンドで初期化をしておく必要があります。

仮想 EEPROM とは内部フラッシュメモリの一部をソフトウェアにて EEPROM のように利用しますものです。仮想 EEPROM に書き込んだデータは、マイコンボードの電源を落としても保持されます。

仮想 EEPROM はフラッシュメモリの 2ページ(1024 バイト×2) を利用しています。

フラッシュメモリの書込み回数の制約を考慮し、同じアドレスへの繰り返し書き込みに対して書き 込み位置を分散しています。

保存可能なデータ量は最大で250ワード(500バイト)となります。

EEPWRITE、EEPREAD コマンドで指定する読書きを指定するアドレスは実際にはアドレスではなく、検索キーです。0~32767の間に任意で利用出来ます。

検索キー数はデータ保存量が250ワードの制約から、250以内に抑える必要あります。仮想EEPROMの機構から数十以内に抑えることを推奨します。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

EEPROM out size : 書き込み容量を超えた

EEPROM bad FLASH : フラッシュメモリの異常が発生した

### ■ 利用例

OK

EEPWRITE 1234, 123 OK ?EEPREAD(1234) 123

#### 7.64 EEPREAD 仮想 EEPROM のからのデータ読み込み (数値関数)

#### ■ 書式

EEPREAD(アドレス)

#### ■ 引数

アドレス : 仮想 EEPROM 内アドレス 0 ~ 32767

#### ■ 戻り値

読みだした 2 バイトデータ (-32767~32767, \$0000~\$FFFF)

## ■ 説明

仮想 EEPROM の指定したアドレスから 2 バイトデータを取得します。 があります。

EEPWRITE、EEPREAD コマンドで指定する読書きを指定するアドレスは実際にはアドレスではなく、検索キーです。0~32767の間に任意で利用出来ます。

指定したアドレスを検索キーとして該当するデータを返します。該当するデータない場合は 0 を返します。

## ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行ったEEPROM bad FLASH: フラッシュメモリの異常が発生した

#### ■ 利用例

EEPWRITE 1234, 123 OK ?EEPREAD(1234) 123 OK

#### 7.65 PSET 点の描画 (一般コマンド)

#### ■ 書式

PSET 横座標, 縦座標, 色

#### ■ 引数

横座標 : 0 ~ GW-1 (注意 最大値は利用環境により異なる場合があります) 縦座標 : 0 ~ GH-1 (注意 最大値は利用環境により異なる場合があります)

色 : 0 黒 、1 白 、2 反転

※GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。

この定数は、画面解像度に設定等により変わります。

## ■ 説明

指定したグラフィク座標に指定した色の点を描画します。

色に2をした場合は、座標位置の色を反転した点を描画します。

本コマンドは NTSC 出力版またはグラフィク表示デバイス版にて利用可能です。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

## ■ 利用例

半径 100 ドット内に点を描画する

10 CLS

20 FOR N=0 TO 30000

30 IF V!=99 Y=Y+V:V=V+1


40 X=RND(200)-100

50 Y=RND(200)-100

60 IF X\*X+Y\*Y<10000 PSET X+110,Y+108,1

70 NEXT N

80 GOTO 80



#### 7.66 LINE 直線の描画(一般コマンド)

#### ■ 書式

LINE 横座標 1, 縦座標 1, 横座標 2, 縦座標 2,色

#### ■ 引数

横座標 1 : 0 ~ GW-1 縦座標 1 : 0 ~ GH-1 横座標 2 : 0 ~ GW-1 縦座標 2 : 0 ~ GH-1

色 : 0 黒 、1 白 、2 反転

※GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。

この定数は、画面解像度に設定等により変わります。

#### ■ 説明

指定したグラフィク座標(横座標 1,縦座標 1)と(横座標 2,縦座標 2)間を結ぶ直線を指定した色で描画 します。色に 2 をした場合は、座標位置の色を反転した点を描画します。

本コマンドは NTSC 出力版またはグラフィク表示デバイス版にて利用可能です。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

## ■ 利用例

直線を描画する


10 CLS

20 FOR X=0 TO 223 STEP 4

30 LINE X,0,223-X,215,1

40 NEXT X

50 GOTO 50



#### 7.67 RECT 矩形の描画 (一般コマンド)

#### ■ 書式

RECT x1, y1,x2, y2, 色, モード

#### ■ 引数

x1: 左上横座標 0 ~ GW-1 (最大値は環境により可変)y1: 左上縦座標 0 ~ GH-1 (最大値は環境により可変)x2: 右下横座標 0 ~ GW-1 (最大値は環境により可変)y2: 右下縦座標 0 ~ GH-1 (最大値は環境により可変)

色 : 0 黒 、1 白 、2 反転

モード: 0 塗りつぶしなし,0以外 塗りつぶしあり

※GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。 この定数は、画面解像度に設定等により変わります。

### ■ 説明

指定した位置に指定した色で矩形を描画します。

モードに0を指定した場合は線のみを描画します。0以外を指定した場合は白で塗りつぶします。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

Illegal value : 指定した数値が不当である

## ■ 利用例

矩形を描画する


10 CLS

20 RECT 50,50,70,90,1,1

30 RECT 60,60,80,80,2,0

40 RECT 80,80,110,110,1,0

50 GOTO 50



#### 7.68 CIRCLE 円の描画 (一般コマンド)

#### ■ 書式

CIRCLE 横中心座標, 縦中心座標, 半径, 色, モード

## ■ 引数

横中心座標:0 ~ GW-1縦中心座標:0 ~ GH-1半径:1 ~ GW-1

色 : 0 黒 、1 白 、2 反転

モード: 0 塗りつぶしなし,0以外 塗りつぶしあり

※GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。

この定数は、画面解像度に設定等により変わります。

#### ■ 説明

指定した中心座標(横中心座標,縦中心座標)を起点に指定した半径の円を色で描画します。 モードに 0 を指定した場合は線のみを描画します。0 以外を指定した場合は指定した色で塗りつぶします。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

#### ■ 利用例

円を描画する

10 CLS

20 FOR Y=0 TO 215 STEP 40


30 FOR X=0 TO 223 STEP 40

40 CIRCLE X,Y,30,1,0

50 NEXT X

60 NEXT Y

70 GOTO 70



7.69 BITMAP ビットマップ画像の描画(一般コマンド)

#### ■ 書式

BITMAP 横座標、縦座標、アドレス、インデックス、幅、高さ [,倍率]

#### ■ 引数

横座標 : 0 ~ GW-1 縦座標 : 0 ~ GH-1

アドレス:数値(任意の相対アドレス)

インデックス: 0 ~ 32767幅: 1 ~ GW-1高さ: 1 ~ GH-1倍率: 1 ~ 8

GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。 この定数は、画面解像度に設定等により変わります。

#### ■ 説明

指定した座標にビットマップ画像を描画します。

アドレスに描画対象の画像データの先頭格納アドレスを指定します。

インデックスには画像データの先頭格納の格納位置を指定します。

(幅 + 7)/8+ 高さ \* インデックス の計算にて参照する画像格納位置を移動します。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

## ■ 利用例

インベーダーのキャラクタを表示する

- 10 CLS
- 20 POKE MEM+0,\$10,\$40,\$48,\$90,\$5F,\$D0,\$77,\$70
- 30 POKE MEM+8,\$3F,\$E0,\$1F,\$C0,\$10,\$40,\$20,\$20
- 40 POKE MEM+16,\$10,\$40,\$08,\$80,\$1F,\$C0,\$37,\$60
- 50 POKE MEM+24,\$7F,\$F0,\$5F,\$D0,\$50,\$50,\$0D,\$80
- 60 FOR A=0 TO 20
- 70 BITMAP A,0,MEM,A%2,12,8,4
- 80 WAIT 200
- 90 NEXT A



#### 7.70 GPRINT 文字列の描画(一般コマンド)

#### ■ 書式

GPRINT 横座標,縦座標

GPRINT 横座標,縱座標文字列|数值

GPRINT 横座標,縱座標文字列|数值;

GPRINT 横座標,縱座標文字列|数值;文字列|数值;…

GPRINT 横座標,縱座標文字列|数值;文字列|数值;…;

GPRINT 横座標,縱座標#桁数,文字列|数值;文字列|数值;…;

※ | はいずれのうち1つを示す

※ …は可変指定を示す

※;は連結指定、カンマ','も可能

## ■ 引数

横座標 : 横描画位置 0 ~ GW-1(通常は 223) 縦座標 : 縦描画位置 0 ~ GH-1(通常は 215)

文字列: 文字列定数または文字列関数

数値 : 数値定数、変数、配列変数、または数値関数、式

連結指定 : セミコロン';' または カンマ';'

文末に付けると改行抑制される

桁数: #数値 または #-数値 の形式で指定する

例:#3 、 #-3

### ■ 説明

指定したグラフィック座標にグラフィックとして文字列を描画します。

表示要素である文字列、数値は連結指定文字のセミコロン';'、またはカンマ','にて連結して表示することが出来ます。

複数の数値を連結して表示した場合、デフォルトでは数値と数値の間は空白が入らない詰めた状態で表示します。#数値にて桁数を指定することで、任意の桁数(指定桁に満たない場合は空白を入れる)にて等間隔で表示します。数値の前に・(マイナス)を付加した場合、空白をではなく、0(零)で不足桁を補います。

PRINT 1;":";2;":";3 桁数指定なし

1:2:3

PRINT #2,1;":";2;":";3 桁数指定あり(空白文字で補間)

1: 2: 3

PRINT #-2,1;":";2;":";3 桁数指定あり(0で補間)

01:02:03

## ■ エラーメッセージ

Syntax error: 文法エラー、書式と異なる利用を行った、引数に数値以外を指定したOverflow: 指定した数値が-32767 ~ 32767 を超えている

## ■ 利用例

## 時刻を表示する

10 GETTIME A,B,C

20 GPRINT 0,0,#-2,A;":";B;":";C

30 WAIT 1000

40 GOTO 20

## 実行結果

10:52:00

7.71 CSCROLL キャラクタ画面スクロール(一般コマンド)

#### ■ 書式


CSCROLL x1,y1,x2,y2,方向

#### ■ 引数

x1: 左上横座標 0 ~ CW-1 (最大値は利用フォントにより可変)y1: 左上縦座標 0 ~ CH-1 (最大値は利用フォントにより可変)x2: 右下横座標 0 ~ CW-1 (最大値は利用フォントにより可変)y2: 右下縦座標 0 ~ CH-1 (最大値は利用フォントにより可変)方向: UP(0): 上、DOWN(1): 下、RIGHT(2): 右、LEFT(3): 左※方向は定数またはカッコ内の数値の指定が可能

#### ■ 説明

画面の指定範囲でテキストを1文字単位でスクロールします。



画面全体をスクロールしたい場合は、

CSCROLL~0,0,CW-1,CH-1,LEFT

のように、画面の左上の位置と画面右下の位置を指定します。

指定した範囲をスクロールしたい場合は、

CSCROLL 3,3,13,5,LEFT

のように範囲を指定します。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

Illegal value : 指定した数値が不当である

## ■ 利用例

文字をスクロールする

**10 CLS** 

20 LOCATE 10,10:?"Hello,TinyBASIC"

30 FOR I=0 TO 15

40 CSCROLL 10,10,24,10,LEFT

50 WAIT 500

60 NEXT I

7.72 GSCROLL グラフィックスクロール(一般コマンド)

#### ■ 書式

GSCROLL x1, y1, x2, y2,方向

#### ■ 引数

x1: 左上横座標 0 ~ GW-1 (最大値は環境により可変)y1: 左上縦座標 0 ~ GH-1 (最大値は環境により可変)x2: 右下横座標 0 ~ GW-1 (最大値は環境により可変)y2: 右下縦座標 0 ~ GH-1 (最大値は環境により可変)

方向: UP(0): 上、 DOWN(1): 下、 RIGHT(2): 右、 LEFT(3): 左

※方向は定数またはカッコ内の数値の指定が可能

※GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。 この定数は、画面解像度に設定等により変わります。

#### ■ 説明

画面上の指定範囲表示内容を1ドット単位でスクロールします。



右上横方向 x1 は8の倍数で指定して下さい(バイト境界指定)。 左下横座標 x2 は8の倍数-1 で指定指定して下さい(バイト境界指定)。 上記以外の数値を指定した場合、指定数値をバイト境界に補正します。

画面全体を左スクロールしたい場合は、 GSCROLL 0, 0,GW-1,GH-1, LEFT のように、画面全体の領域を指定します。

指定した範囲をスクロールしたい場合は、 GSCROLL 16,30,207,100, LEFT のように範囲を指定します。 (注意) GSCROLL でスクロールした場合、見かけ上のテキストとテキスト表示管理の情報にずれが生じます。VPEEKによる表示位置の文字コード取得は正しい値を返しません。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

Illegal value : 指定した数値が不当である

## ■ 利用例

簡易オシロスコープ

10 'oscilloscope

20 CLS

30 GPIO PB01, ANALOG

40 "LOOP"

50 R=ANA(PB01)

60 Q=R/20

70 LOCATE 0,0:?R;" "

80 PSET 223,221-Q,1

90 WAIT 50

100 GSCROLL 0,8,223,215,LEFT

110 GOTO "LOOP"

#### 7.73 GPEEK 画面上の指定位置ピクセルの参照(数値関数)

#### ■ 書式

GPEEK(横位置 , 縦位置)

#### ■ 引数

横位置: 0~GW-1 縦位置: 0~GH-1

※GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。 環境等により異なります。

## ■ 戻り値

指定位置に表示されているピクセルの色 0 または 1

#### ■ 説明

画面上の指定位置に表示されているピクセルを参照します。 引数の指定位置が範囲外の場合は 0 を返します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

#### ■ 利用例

画面上に表示しているドットに有無を2進数で表示する

- 10 B=0
- 10 FOR I=0 TO 7
- 20 B=(B<<1)+GPEEK(I,0)
- 30 NEXT I
- 40 PRINT BIN\$(B,8)
- 50 END

#### 7.74 GINP 指定領域のピクセルの有無判定(数値関数)

#### ■ 書式

GINP(横位置,縦位置,高さ,幅,色)

#### ■ 引数

横位置 : 0~GW·1 縦位置 : 0~GH·1 幅 :1 ~ GW·1 高さ :1 ~ GH·1 色 :0 黒 、1 白

※GW、GH はグラフィック画面の横ドット数、縦ドット数を示す定数です。


環境等により異なります。

## ■ 戻り値

1:指定領域内(境界含む)に指定した色のピクセルが存在する 0:指定領域内(境界含む)に指定した色のピクセルが存在しない

## ■ 説明

指定した領域内の指定した色のピクセルが存在するをチェックし、その結果を返します。



## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32767 ~ 32767 を超えている

'(' or ')' expected : '(' または ')'が無い

## ■ 利用例

画面上に表示しているドットに有無を2進数で表示する

10 B=0

10 FOR I=0 TO 7

20 B=(B<<1)+GPEEK(I,0)

30 NEXT I

40 PRINT BIN\$(B,8)

50 END

#### 7.75 TONE 単音出力(一般コマンド)

#### ■ 書式

TONE 周波数

TONE 周波数,出力期間

## ■ 引数

周波数 : 0 ~ 32767 (Hz) 0 の場合は消音

出力期間 : 0 ~ 32767(ミリ秒) 0の場合は、継続再生

#### ■ 説明

PB9 ピンより、指定した周波数のパルス出力(デューティ比 50%)を行います。

PB9 ピンに圧電スピーカー (圧電サウンダ) を接続すること音を出すことが出来ます。

出力期間の指定がある場合は、その期間パスルを出力します(ミリ秒単位)。

出力期間の指定がある場合、出力完了待ちを行います。

出力期間の指定がない場合は、NOTONE コマンドで停止指示をするまでパスルを出力し続けます。 (TONE 0 は NOTONE と等価です)

辛陛		田	油	米石	54	<u>,</u>	丰
- 123	•		:10	<i>₩</i>	XVI	hi's	オセ

	7.	#1	レ	レ#	ııı	ファ	ファ井	ソ	ソ#	ラ	ラ#	シ
1	33	35	37	39	41	44	46	49	52	55	58	62
2	65	69	73	78	82	87	93	98	104	110	117	123
3	131	139	147	156	165	175	185	196	208	220	233	247
4	262	277	294	311	330	349	370	392	415	440	466	494
5	523	554	587	622	659	698	740	784	831	880	932	988
6	1047	1109	1175	1245	1319	1397	1480	1568	1661	1760	1865	1976
7	2093	2217	2349	2489	2637	2794	2960	3136	3322	3520	3729	3951
8	4186	4435	4699	4978	5274	5588	5920	6272	6643	7040	7459	7902

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow: 指定した数値が-32767 ~ 32767 を超えているIllegal value: ピン番号、モードに範囲外の値を指定した

#### ■ 利用例

スペースキーを押したら音を鳴らす 10 IF INKEY() = 32 TONE 800,50 20 GOTO 10

## 7.76 NOTONE 単音出力停止(一般コマンド)

## ■ 書式

NOTONE

## ■ 引数

なし

## ■ 説明

TONE コマンドによるパスル出力を停止します。

## ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

## ■ 利用例

10 TONE 400

20 WAIT 200

30 NOTONE

# 9. 応用プログラム

9.1 I2C インタフェースの利用例 I2C バスを利用するには、 $2.2 \mathrm{k} \sim 10 \mathrm{k} \, \Omega$  程度でプルアップする必要があります。 (以降作成中)

- 9.2 アナログ入力の利用例
- 9.3 SHIFTOUT コマンドを使った I/O 制御 (以降作成中)