

豊四季タイニーBASIC for Arduino STM32 V0.3

リファレンスマニュアル RV0.2

作成日 2017 年 4 月 4 日

作成者 たま吉

## 目次

1	概要 .....	3
1.1	はじめに.....	3
1.2	システム構成 .....	3
1.3	コンパイル方法.....	3
2	文法 .....	3
2.1	コマンド.....	3
2.2	式.....	3
2.3	関数.....	3
2.4	制御文 .....	3
2.5	定数.....	4
3	コマンド・関数の説明 .....	5
3.1	RENUM 行番号の振り直し（コマンド） .....	5
3.2	CLS 画面表示内容の全消去（コマンド） .....	6
3.3	LOCATE カーソルの移動（コマンド） .....	7
3.4	COLOR 文字色の設定（コマンド） .....	8
3.5	ATTR 文字表示属性の設定（コマンド） .....	9
3.6	WAIT 時間待ち（コマンド） .....	10
3.7	VPEEK 画面指定位置の文字コード参照（関数） .....	11
3.8	CHR\$ 画文字コードから文字への変換（関数） .....	12
3.9	ASC 文字から文字コードへの変換（関数） .....	13
3.10	HEX\$ 数値から 16 進数文字列への変換（関数） .....	14
3.11	BIN\$ 数値から 2 進数文字列への変換（関数） .....	15
3.12	TICK 起動からの経過時間取得（関数） .....	16
3.13	RESETTICK 起動からの経過時間カウンタのリセット（コマンド） .....	17
3.14	GPIO GPIO 機能設定（コマンド） .....	18
3.15	OUT デジタル出力（コマンド） .....	19
3.16	IN デジタル入力（関数） .....	20
3.17	ANA アナログ入力（関数） .....	21
3.18	SHIFTOUT デジタルシフトアウト出力（コマンド） .....	22
3.19	SHIFTIN デジタルシフトアウト入力（関数） .....	23
3.20	INKEY キー入力の読み取り（関数） .....	24
3.21	DATE 現在時刻の表示（コマンド） .....	25
3.22	SETDATE 時刻の設定（コマンド） .....	26
3.23	GETDATE 日付の取得（コマンド） .....	27

3.24	GETTIME	時刻の取得（コマンド） .....	28
3.25	PEEK	指定アドレスの値参照（関数） .....	29
3.26	POKE	指定アドレスへのデータ書き込み（コマンド） .....	30
3.27	SAVE	内部フラッシュメモリへのプログラム保存（コマンド） .....	31
3.28	LOAD	内部フラッシュメモリからプログラムを読み込む（コマンド） .....	32
3.29	I2CW	I2C スレーブデバイスへのデータ送信（関数） .....	33
3.30	I2CR	I2C スレーブデバイスからのデータ受信（関数） .....	35

## 1 概要

- 1.1 はじめに
- 1.2 システム構成
- 1.3 コンパイル方法

## 2 文法

- 2.1 コマンド
- 2.2 式
- 2.3 関数
- 2.4 制御文

## 2.5 定数

「Tiny BASIC for Arduino STM32」では次の定数が利用出来ます。  
定数はコマンドの引数や式の中で数値関数と同等に利用出来ます。

### ■ 1 ビット入出力値

HIGH, LOW

HIGH は 1, LOW は 0 が割り当てられています。  
各コマンドの引数、IF 文、式にて利用出来ます。

### ■ メモリ領域先頭アドレス定数

MEM, VRAM, VAR, ARRAY

SRAM の先頭アドレスからの相対アドレスを参照するための定数です。  
詳細は次の通りです。

MEM	: ユーザーワーク領域	サイズ 1024 バイト
VRAM	: 画面表示用メモリ (80×25)	サイズ 2048 バイト
VAR	: 変数領域 (A～Z)	サイズ 52 バイト
ARRAY	: 配列変数領域 (@(0)～@(99))	サイズ 200 バイト

PEEK、POKE、I2CW、I2CR 等の SRAM メモリを利用するコマンドで利用します。

### ■ ピン番号定数

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08  
PA09, PA10, PA11, PA12, PA13, PA14, PA15  
PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08  
PB09, PB10, PB11, PB12, PB13, PB14, PB15  
PC13, PC14, PC15

GPIO、OUT、IN、ANA、SHIFTOUT、SHIF TIN コマンド・関数のピン番号の指定に利用します。  
各ピン番号定数に実際のピン番号 0 ～ 34 が割り当てられています。

### ■ GPIO モード設定定数

OUTPUT\_OD, OUTPUT, INPUT\_PU, INPUT\_PD, ANALOG, INPUT\_FL  
GPIO コマンドのモード設定を行うための定数です。

### ■ ビット方向定数

LSB, MSB

ビット送信等で上位、下位を指定するための定数です。  
SHIF TIN, SHFITOUT コマンドで利用します。

### 3 コマンド・関数の説明

#### 3.1 RENUM 行番号の振り直し（コマンド）

##### ■ 書式

RENUM

RENUM 開始行番号

RENUM 開始行番号, 増分

##### ■ 引数

開始番号： 振り直しをする新しい行番号の開始番号 （1 ～ 32767）

増 分： 行番号の増分 （1 ～ 32767）

##### ■ 説明

プログラム全体の行番号を指定した条件にて振り直します。

行番号を指定した開始番号から指定した増分で振り直します。

行番号は 1～32767 の範囲まで振ることが出来ます。

引数を省略した場合は、行番号を 10 行から 10 間隔で振り直します。

開始番号だけを指定した場合は、指定した開始番号から 10 間隔再振り直します。

開始番号と増分を指定した場合は、指定した開始番号から指定した増分で振り直します。

振り直しにおいて、GOTO 文、GOSUB 文のとび先の行番号も正しく更新されます。

開始行番号には変数、式の指定は行えません。必ず数値を指定して下さい。

##### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った、引数に数値以外を指定した
Illegal value	: 振り直しをする新しい行番号が有効範囲を超えている
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

##### ■ 利用例

```
LIST
10 I=0
20 PRINT "*";
30 I=I+1:IF I<10 GOTO 20
40 PRINT
50 END
OK
RENUM 100,10
OK
LIST
100 I=0
110 PRINT "*";
120 I=I+1:IF I<10 GOTO 110
130 PRINT
140 END
```

## 3.2 CLS 画面表示内容の全消去（コマンド）

### ■ 書式

CLS

### ■ 引数

なし

### ■ 説明

画面上に表示している内容を全て消します。

行番号を指定した開始番号から指定した増分で振り直します。

カーソルは画面の先頭左上に移動します。

（注意）直前に COLOR コマンドにて文字色、背景色の指定を行っている場合、画面全体に背景色が適用されます。

### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

### ■ 利用例

CLS

OK

### 3.3 LOCATE カーソルの移動（コマンド）

#### ■ 書式

LOCATE 横位置, 縦位置

#### ■ 引数

横位置：画面上の横位置 0 ～ 79（コンソール出力の場合）

縦位置：画面上の縦位置 0 ～ 24（コンソール出力の場合）

#### ■ 説明

カーソルを指定した位置に移動します。

縦横位置それぞれの指定に 0 以下の数値を指定した場合、それぞれの位置は 0 となります。

横位置に 79 を超える数値を指定した場合、横位置は 79 となります。縦位置に 24 を超える数値を指定した場合、縦位置は 24 となります。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

画面の指定位置にメッセージを表示する

```
10 CLS
20 LOCATE 5,5:PRINT "Hello,world."
30 LOCATE 6,6:PRINT "TinyBASIC"
40 LOCATE 7,7:PRINT "Thank you."
```

実行結果

```
Hello,world.
TinyBASIC
Thank you.
```



### 3.4 COLOR 文字色の設定（コマンド）

#### ■ 書式

COLOR 文字色

COLOR 文字色, 背景色

#### ■ 引数

文字色： 色コード 0～9

背景色： 色コード 0～9

#### ■ 説明

文字色の設定を行います。指定した色は以降の文字表示に反映されます。

文字色、背景色で指定する色コードに対する色は次の表の通りです。

表 1 色コード

色コード	色
0	黒
1	赤
2	緑
3	茶
4	青
5	マゼンタ
6	シアン
7	白(デフォルト)
8	黄

（注意）利用するターミナルソフトにより色が正しく表示されない場合があります。

属性指定との併用では正しく表示されない場合があります。

画面を[CTRL-R]、[Page UP]、[Page Down]キーにて再表示した場合、色情報は欠落します。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った  
 Illegal value : 色コードに範囲外の値を指定した  
 Overflow : 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

ランダムな色で\*を表示する

```
10 FOR I=0 TO 10
20 FOR J=0 TO 10
30 COLOR RND(8): ? "*";
35 WAIT 100
40 NEXT J
50 ?
60 NEXT I
```

### 3.5 ATTR 文字表示属性の設定（コマンド）

#### ■ 書式

ATTR 属性

#### ■ 引数

属性： 属性コード 0 ～ 4

#### ■ 説明

文字の表示属性を設定します。指定した表示属性は以降の文字表示に反映されます。  
属性に指定する属性コードは次の表の通りです。

表 2 属性コード

属性コード	機能
0	標準(デフォルト)
1	下線
2	反転
3	ブリンク
4	ボールド

(注意) 利用するターミナルソフトにより色が正しく表示されない場合があります。  
色指定との併用指定を行った場合、正しく表示されない場合があります。  
画面を[CTRL-R]、[Page UP]、[Page Down]キーにて再表示した場合、  
色情報は欠落します。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った  
Illegal value : 属性コードに範囲外の値を指定した  
Overflow : 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

Hello,world を反転表示する

```
10 CLS
20 LOCATE 5,5
30 ATTR 2:? "Hello,world"
40 ATTR 0
```

### 3.6 WAIT 時間待ち（コマンド）

#### ■ 書式

WAIT 待ち時間(ミリ秒)

#### ■ 引数

待ち時間： 0 ～ 32767 （単位 ミリ秒）

#### ■ 説明

引数で指定した時間(ミリ秒単位)、時間待ち(ウェイト)を行います。

最大で 32767 ミリ秒（32.8 秒）の時間待ちが可能です。

長い時間待ちを行う必要がある場合は、TICK0や GETTIME を使った方法を検討してください。

（注意）時間待ち中はキー操作によるプログラム中断を行うことは出来ません。

短い時間の指定にてループ処理を行う等の対策を行ってください。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 待ち時間に範囲外の値を指定した
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

画面上の指定位置に時刻を 1 秒間隔で更新表示する

```
10 SETDATE 2017,4,1,12,0,0
20 CLS
30 LOCATE 5,5
40 DATE
50 WAIT 1000
60 GOTO 30
```

### 3.7 VPEEK 画面指定位置の文字コード参照（関数）

#### ■ 書式

VPEEK(横位置 , 縦位置)

#### ■ 引数

横位置 : 0～79

縦位置 : 0～24

#### ■ 戻り値

指定位置に表示されている文字の文字コード (0 ～ 255)

#### ■ 説明

画面上の指定位置に表示されている文字の文字コードを取得します。

引数の指定位置が範囲外の場合は 0 を返します。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Overflow : 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

画面最上位の行に表示されている文字を表示する

```
10 FOR I=0 TO 79
20 C=VPEEK(I,0)
30 PRINT CHR$(C);
40 NEXT I
50 PRINT
```

### 3.8 CHR\$ 画文字コードから文字への変換（関数）

#### ■ 書式

CHR\$(文字コード)

#### ■ 引数

文字コード： 0～127, 160～255

#### ■ 戻り値

指定位置に表示されている文字の文字コード（0 ～ 255）

#### ■ 説明

指定した文字コードに対応する文字を返します。

本関数は PRINT の引数として利用可能です。

範囲外の値を指定した場合は空白文字(" ")を返します。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
'(' or ')' expected	: '(' または ')'が無い
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

アルファベット A～Z を表示する

```

10 C=ASC("A")
20 FOR I=0 TO 25
30 PRINT CHR$(C+I);
40 NEXT I
50 PRINT

run
ABCDEFGHIJKLMNOPQRSTUVWXYZ
OK
    
```

### 3.9 ASC 文字から文字コードへの変換（関数）

#### ■ 書式

ASC(文字)

#### ■ 引数

文字 : “文字”

“A” の形式とし、ダブルクオテーションで文字を囲みます

#### ■ 戻り値

指定文字に対応する文字コード (0 ~ 255)

#### ■ 説明

指定した文字に対応する文字コードを返します.

指定した文字列が 1 文字で無い場合は、エラーとなります。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

Illegal value : 文字数が不当

'(' or ')' expected : '(' または ')'がない

#### ■ 利用例

アルファベット A~Z を表示する

```
10 C=ASC("A")
20 FOR I=0 TO 25
30 PRINT CHR$(C+I);
40 NEXT I
50 PRINT

run
ABCDEFGHIJKLMNOPQRSTUVWXYZ
OK
```

### 3.10 HEX\$ 数値から 16 進数文字列への変換（関数）

#### ■ 書式

HEX\$(数値)

HEX\$(数値 , 桁数)

#### ■ 引数

数値： 変換対象の整数値(-32768 ~ 32767)

桁数： 出力桁数(0 ~ 4)

#### ■ 戻り値

16 進数文字列(1 桁~4 桁)

#### ■ 説明

指定した数値を 16 進数文字列に変換します。

PRINT 文の引数にて利用可能です。

桁数を指定した場合は数値が指定した桁数に満たない場合は、0 で桁を補います。

指定した桁数を超える場合は数値の桁数を優先します。

桁数をしない場合は、先頭の 0 は付加されません。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
'(' or ')' expected	: '(' または ')'がない
Illegal value	: 桁数の値が正しくない
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

#### ■ 利用例

### 3.11 BIN\$ 数値から 2 進数文字列への変換（関数）

#### ■ 書式

BIN\$(数値)

BIN\$(数値 , 桁数)

#### ■ 引数

数値： 変換対象の整数値( -32768 ～ 32767 )

桁数： 出力桁数( 0 ～ 16)

#### ■ 戻り値

2 進数文字列(1 桁～16 桁)

#### ■ 説明

指定した数値を 2 進数文字列に変換します。

PRINT 文の引数にて利用可能です。

桁数を指定した場合は数値が指定した桁数に満たない場合は、0 で桁を補います。

指定した桁数を超える場合は数値の桁数を優先します。

桁数をしない場合は、先頭の 0 は付加されません。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
'(' or ')' expected	: '(' または ')' がない
Illegal value	: 桁数の値が正しくない
Overflow	: 指定した数値が -32768 ～ 32767 を超えている

#### ■ 利用例

2 進数で変数の内容を表示する

```
10 A=1234:B=1
20 PRINT BIN$(A)
30 PRINT BIN$(B)
40 PRINT BIN$(A,4)
50 PRINT BIN$(B,4)
```

```
run
10011010010
1
10011010010
0001
OK
```



### 3.12 TICK 起動からの経過時間取得（関数）

#### ■ 書式

TICK()

TICK(モード)

#### ■ 引数

モード： LOW (または 0) 下位 15 ビットの値取得

HIGH (または 1) 上位 15 ビットの値取得

#### ■ 戻り値

モード LOW 時： 0 ～ 32767 (0.1 秒単位) 約 54.6 分迄

モード HIGH 時： 0 ～ 32767 (3276.8 秒単位) 約 54.6 分 × 取得数値

#### ■ 説明

起動からの経過時間を返します。

モード指定無、または LOW 指定の場合は 0.1 秒単位の経過時間を返します。

モード指定が HIGH 場合は上位のビットを返します。

LOW の桁は約 54.6 分で桁溢れして 0 に戻ります。桁溢れの発生を判定する場合は HIGH 指定にて上位ビットを参照して下さい。

秒単であれば、内蔵 RTC を使う GETTIME を使う方法もあります。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

Overflow : 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

5 秒間時間待ちループを行う

```
10 RESETTICK
20 IF TICK(<50 GOTO 20
30 PRINT "5sec passed"
```

### 3.13 RESETTICK 起動からの経過時間カウンタのリセット（コマンド）

#### ■ 書式

RESETTICK

#### ■ 引数

なし

#### ■ 説明

起動からの経過時間カウンタをリセット（0 を設定）します。

TICK0関数使う前に経過時間カウンタをリセットすることで 0 からのカウントを行うことが出来ます。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

#### ■ 利用例

5 秒間時間待ちループを行う

```
10 RESETTICK
20 IF TICK(<)<50 GOTO 20
30 PRINT "5sec passed"
```

### 3.14 GPIO GPIO 機能設定（コマンド）

#### ■ 書式

GPIO ピン番号, モード

#### ■ 引数

ピン番号 : 0 ~ 34

ピン番号は数値の他に次のピン名（定数）での指定も可能です。

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,  
PA09, PA10, PA11, PA12, PA13, PA14, PA15,  
PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,  
PB09, PB10, PB11, PB12, PB13, PB14, PB15,  
PC13, PC14, PC15

モード : 次の定数

OUTPUT\_OD : デジタル出力（オープンドレイン）  
OUTPUT : デジタル出力  
INPUT\_FL : デジタル入力（フロート状態 : Arduino の INPUT 指定と同じ）  
INPUT\_PU : デジタル入力（内部プルアップ抵抗有効）  
INPUT\_PD : デジタル（内部プルダウン抵抗有効）  
ANALOG : アナログ入力

#### ■ 説明

ボード上の GPIO ピンの機能設定を行います。

Arduino の pinMode() に相当します。

GPIO ピンを使って信号の入出力を行う場合は、必ず本コマンドによる設定が必要です。

（注意）ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。

ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った  
Illegal value : ピン番号、モードに範囲外の値を指定した  
Overflow : 指定した数値が -32768 ~ 32767 を超えている

#### ■ 利用例

PB01 ピンからアナログ入力値を読み取り、その値を画面に随時表示します。

```
10 CLS
20 GPIO PB01,ANALOG
30 A=ANA(PB01)
40 LOCATE 5,5: ? A; "    "
50 GOTO 30
```

### 3.15 OUT デジタル出力（コマンド）

#### ■ 書式

OUT ピン番号, 出力値

#### ■ 引数

ピン番号 : 0 ~ 34

ピン番号は数値の他に次のピン名（定数）での指定も可能です。

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,  
PA09, PA10, PA11, PA12, PA13, PA14, PA15,  
PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,  
PB09, PB10, PB11, PB12, PB13, PB14, PB15,  
PC13, PC14, PC15

出力値 :

LOW または 0 : 0V を出力する

HIGH or 0 以外の値 : 3.3V を出力する

#### ■ 説明

指定ピンから、指定した出力を行います。

出力を行う場合は事前に GPIO コマンドによる機能設定（出力設定）が必要です。

（注意）ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。

ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

#### ■ 利用例

Blue Pill ボードの搭載 LED（PC13 ピン）を点滅させます。

```
10 P=PC13
20 GPIO P,OUTPUT
30 OUT P,HIGH
40 FOR I=1 TO 10
50 OUT P,LOW
60 WAIT 300
70 OUT P,HIGH
80 WAIT 300
90 NEXT I
```

### 3.16 IN デジタル入力（関数）

#### ■ 書式

IN(ピン番号)

#### ■ 引数

ピン番号 : 0 ~ 34

ピン番号は数値の他に次のピン名（定数）での指定も可能です。

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,  
PA09, PA10, PA11, PA12, PA13, PA14, PA15,  
PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,  
PB09, PB10, PB11, PB12, PB13, PB14, PB15,  
PC13, PC14, PC15

#### ■ 戻り値

取得した値 0(LOW) または 1(HIGH)

#### ■ 説明

指定ピンの入力値を読み取り、その値を返します。

入力を行う場合は事前に GPIO コマンドによる機能設定（入力設定）が必要です。

（注意）ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。

ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Overflow	: 指定した数値が-32768 ~ 32767 を超えている
'(' or ')' expected	: '(' または ')'がない

#### ■ 利用例

### 3.17 ANA アナログ入力（関数）

#### ■ 書式

ANA(ピン番号)

#### ■ 引数

ピン番号： 0 ～ 7, 16, 17 または以下の定数

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07,  
PB00, PB0

#### ■ 戻り値

取得した値 0～4095(12 ビット)

#### ■ 説明

指定ピンのアナログ入力値を読み取り、その値を返します。

アナログ入力を行う場合は事前に GPIO コマンドによる機能設定（アナログ入力）が必要です。

（注意）ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。

ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Overflow	: 指定した数値が-32768 ～ 32767 を超えている
'(' or ')' expected	: '(' または ')'がない

#### ■ 利用例

PB01 ピンからアナログ入力値を読み取り、その値を画面に随時表示します。

```
10 CLS
20 GPIO PB01,ANALOG
30 A=ANA(PB01)
40 LOCATE 5,5: ? A; "    "
50 GOTO 30
```

### 3.18 SHIFTOUT デジタルシフトアウト出力（コマンド）

#### ■ 書式

SHIFTOUT データピン番号, クロックピン番号, 出力形式, 出力データ

#### ■ 引数

データピン番号 : 0 ~ 34 データを出力するピン

クロックピン番号 : 0 ~ 34 クロックを出力するピン

上記のピン番号は数値の他に次のピン名（定数）での指定も可能です。

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,  
PA09, PA10, PA11, PA12, PA13, PA14, PA15,  
PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,  
PB09, PB10, PB11, PB12, PB13, PB14, PB15,  
PC13, PC14, PC15

出力形式 : 出力するデータの順番を下記にて指定

LSB または 0 : 下位ビットから出力する

MSB または 1 : 上位ビットから出力する

出力データ : 出力するデータ（下位 8 ビットのみ有効）

#### ■ 説明

クロックにて同期を行い、データピンから 1 バイト分のデータを 1 ビットずつ出力します。

Arduino の shiftOut() と同等の動作をします。

データピン、クロックピンは事前に GPIO コマンドによる機能設定（デジタル出力）が必要です。

（注意）ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。

ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Overflow	: 指定した数値が -32768 ~ 32767 を超えている

#### ■ 利用例

### 3.19 SHIFTIN デジタルシフトアウト入力（関数）

#### ■ 書式

SHIFTIN(データピン番号, クロックピン番号, 入力形式)

#### ■ 引数

データピン番号 : 0 ~ 34 データを入力するピン

クロックピン番号 : 0 ~ 34 クロックを出力するピン

上記のピン番号は数値の他に次のピン名（定数）での指定も可能です。

PA00, PA01, PA02, PA03, PA04, PA05, PA06, PA07, PA08,  
PA09, PA10, PA11, PA12, PA13, PA14, PA15,  
PB00, PB01, PB02, PB03, PB04, PB05, PB06, PB07, PB08,  
PB09, PB10, PB11, PB12, PB13, PB14, PB15,  
PC13, PC14, PC15

入 力 形 式 : 入力するデータの順番を下記にて指定

LSB または 0 : 下位ビットから入力する

MSB または 1 : 上位ビットから入力する

#### ■ 戻り値

入力値（1 バイト）

#### ■ 説明

クロックにて同期を行い、データピンから 1 バイト分のデータを 1 ビットずつ入力します。

Arduino の shiftIn() と同等の動作をします。

データピン、クロックピンは事前に GPIO コマンドによる機能設定（デジタル入出力）が必要です。

（注意）ピン番号の指定範囲及び定数は、ST32F103C8T6 での利用を想定したものです。

ST32F103C8T6 以外の MPU にて使う場合は、機能設定できない場合があります。

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: ピン番号、モードに範囲外の値を指定した
Overflow	: 指定した数値が-32768 ~ 32767 を超えている
'(' or ')' expected	: '(' または ')' がない

#### ■ 利用例



### 3.20 INKEY キー入力の読み取り（関数）

#### ■ 書式

INKEY()

#### ■ 引数

なし

#### ■ 戻り値

押したキーの文字コード

キーが押されていない場合は 0

#### ■ 説明

キーボード上の押しているキーの文字コードを取得します。

キーが押されていない場合は、0 を返します。

（注意）[ESC]、[CTRL-C]はプログラム中断用のため、キー入力の読み取りは出来ません。

また、コンソールの制約により読み取れないキーがあります。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

'(' or ')' expected : '(' または ')'がない

#### ■ 利用例

### 3.21 DATE 現在時刻の表示（コマンド）

#### ■ 書式

DATE

#### ■ 引数

なし

#### ■ 説明

内蔵 RTC から現在の時刻を読み、その情報を画面に表示します。

（注意）内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行って下さい。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

#### ■ 利用例

時刻の設定と表示を行う

```
SETDATE 2017,4,1,12,00,00
OK
DATE
2017/04/01 [Sat] 12:00:03
OK
```

## 3.22 SETDATE 時刻の設定（コマンド）

### ■ 書式

SETDATE 年,月,日,時,分,秒

### ■ 引数

年 : 1900 ~ 2036	西暦年 4 桁の整数
月 : 1 ~ 12	整数
日 : 1 ~ 31	整数
時 : 0 ~ 23	整数
分 : 0 ~ 59	整数
秒 : 0 ~ 61	整数（うるう秒考慮）

### ■ 説明

指定した時刻を内蔵 RTC に設定します。

（注意）内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行って下さい。

### ■ エラーメッセージ

Illegal value	: 指定した引数の数値が有効範囲以外
Syntax error	: 文法エラー、書式と異なる利用を行った
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

### ■ 利用例

時刻の設定と表示を行う

```
Y=2017:M=4:D=1:H=12:N=0:S=0
SETDATE Y,M,D,H,N,S
OK
DATE
2017/04/01 [Sat] 12:00:03
OK
```

### 3.23 GETDATE 日付の取得（コマンド）

#### ■ 書式

GETDATE 年格納変数,月格納変数,日格納変数,曜日格納変数

#### ■ 引数

年格納変数：取得した西暦年を格納する変数を指定

月格納変数：取得した月を格納する変数を指定

日格納変数：取得した日を格納する変数を指定

曜日格納変数：取得した曜日コードを格納する変数を指定

#### ■ 説明

内蔵 RTC から日付情報を取得し、その値を指定した変数に格納します。

格納される値を次の通りです。

年格納変数：西暦年 4 桁整数 1900 ~ 2036

月格納変数：1 ~ 12

日格納変数：1 ~ 31

曜日格納変数：曜日コード 0 ~ 6 (0:日 1:月 2:火 3:水 4:木 5:金 6:土)

（注意）内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行ってください。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

#### ■ 利用例

現在の日付を取得する

```
DATE
2017/04/01 [Sat] 12:03:05
OK
GETDATE A,B,C,D
OK
PRINT a;b;c;d
2017416
OK
```

### 3.24 GETTIME 時刻の取得（コマンド）

#### ■ 書式

GETTIME 時格納変数,分格納変数 秒格納変数

#### ■ 引数

時格納変数： 取得した時を格納する変数を指定

分格納変数： 取得した分を格納する変数を指定

秒格納変数： 取得した秒を格納する変数を指定

#### ■ 説明

内蔵 RTC から日付情報を取得し、その値を指定した変数に格納します。

格納される値を次の通りです。

時格納変数： 0 ～ 23 整数

分格納変数： 0 ～ 59 整数

秒格納変数： 0 ～ 61 整数(うるう秒考慮)

（注意）内蔵 RTC 用のバックアップ電池を搭載していないボードでは、時刻情報は電源 OFF により初期化されます。その場合は SETDATE コマンドにて時刻設定を行ってください。

#### ■ エラーメッセージ

Syntax error : 文法エラー、書式と異なる利用を行った

#### ■ 利用例

現在の時間を取得する

```
DATE
2017/04/01 [Sat] 12:10:51
OK
GETTIME A,B,C
OK
PRINT #-2,a;":";b;":";c
12:11:01
OK
```

(補足) PRINT 文の #-2 は数値を 2 桁(0 付き)で表示する指定です。

### 3.25 PEEK 指定アドレスの値参照（関数）

#### ■ 書式

PEEK(相対アドレス)

#### ■ 引数

相対アドレス： 参照を行う SRAM 領域先頭からの相対アドレス(16 ビット)

#### ■ 戻り値

指定した相対アドレス内の 1 バイトデータ（0～255）

#### ■ 説明

SRAM 先頭からの相対アドレスを指定し、格納されている値（1 バイト）を返します。

相対アドレスの指定には次の定数を利用することで有用な領域への参照が簡単に行えます。

MEM	: ユーザーワーク領域	サイズ 1024 バイト
VRAM	: 画面表示用メモリ (80×25)	サイズ 2048 バイト
VAR	: 変数領域 (A～Z)	サイズ 52 バイト
ARRAY	: 配列変数領域(@ (0)～@ (99))	サイズ 200 バイト

ユーザーワーク領域は利用者が自由に利用出来る領域です。

それ以外の領域は BASIC の実行にて利用する領域です。

例として画面上のカーソル位置 X,Y に格納されている文字の参照は次のようになります。

```
C = PEEK(VRAM+Y*80+X)
```

（注意）SRAM の領域外のアドレスを指定した場合、エラーとなります。

領域チェックにはマクロ SRAM\_SIZE の定義値（デフォルト 20480）を利用しています。

デフォルト値よりも大きい容量のマイコンを利用している場合はこの定義値を修正して下さい。

#### エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Out of range value	: SRAM の領域外のアドレスを指定した
'(' or ')' expected	: 括弧の指定が正しくない
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

変数 A から Z をユーザーワーク領域に保存する例

```
10 FOR I=0 TO 51
20 POKE MEM+I,PEEK(VAR+I)
30 NEXT I
```

### 3.26 POKE 指定アドレスへのデータ書き込み（コマンド）

#### ■ 書式

POKE 相対アドレス,データ

POKE 相対アドレス,データ,データ, ... データ （可変個数指定）

#### ■ 引数

相対アドレス： 参照を行う SRAM 領域先頭からの相対アドレス(16 ビット)

データ： 書き込むデータ（下位 8 ビットのみ有効）

#### ■ 説明

SRAM 先頭からの指定した相対アドレスに指定したデータを書き込みます。

相対アドレスの指定には次の定数を利用することで有用な領域への書き込みが簡単に行えます。

MEM	: ユーザーワーク領域	サイズ 1024 バイト
VRAM	: 画面表示用メモリ (80×25)	サイズ 2048 バイト
VAR	: 変数領域 (A～Z)	サイズ 52 バイト
ARRAY	: 配列変数領域(@0)～@99)	サイズ 200 バイト

ユーザーワーク領域は利用者が自由に利用出来る領域です。

それ以外の領域は BASIC の実行にて利用する領域です。

指定した相対アドレスに上記以外の領域を指定した場合はエラーとなります。

例として画面上のカーソル位置 X,Y への文字の書き込みは次のようになります。

```
POKE VRAM+Y*80+X,ASC("A")
```

#### ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Out of range value	: SRAM の領域外のアドレスを指定した
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

変数 A から Z をユーザーワーク領域に保存する例

```
10 FOR I=0 TO 51
20 POKE MEM+I,PEEK(VAR+I)
30 NEXT I
```

### 3.27 SAVE 内部フラッシュメモリへのプログラム保存（コマンド）

#### ■ 書式

SAVE

SAVE プログラム番号

#### ■ 引数

プログラム番号： 0 ～ 3

#### ■ 説明

プログラムをマイコン内のフラッシュメモリに保存します。

プログラムは最大で 4 つ保存可能です。保存先はプログラム番号 0～3 で指定します。

プログラム番号を指定しない場合、プログラム番号に保存します。

プログラム番号の指定には、変数や式の指定は出来ません。

（注意）SAVE コマンドはプログラム内での実行は出来ません。実行しても無視されます。  
直接コマンド入力を行った場合のみ有効です。

#### ■ エラーメッセージ

Syntax error	: 書式と異なる利用を行った、プログラム番号に変数、式を指定した
Illegal value	: プログラム番号の指定が 0～3 の範囲外である
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

プログラム番号 1 にプログラムを保存する

SAVE 1

OK



### 3.28 LOAD 内部フラッシュメモリからプログラムを読み込む（コマンド）

#### ■ 書式

LOAD

LOAD プログラム番号

#### ■ 引数

プログラム番号： 0 ～ 3

#### ■ 説明

マイコン内のフラッシュメモリからプログラムを読み込みます。

プログラム番号 0～3 で指定します。プログラムは最大で 4 つ保存可能です。

プログラム番号の指定しない場合、プログラム番号 0 を読み込みます。

プログラム番号の指定には、変数や式の指定は出来ません。

（注意）LOAD コマンドはプログラム内での実行は出来ません。実行しても無視されます。  
直接コマンド入力を行った場合のみ有効です。

#### ■ エラーメッセージ

Syntax error	: 書式と異なる利用を行った、プログラム番号に変数、式を指定した
Illegal value	: プログラム番号の指定が 0～3 の範囲外である
Program not found	: 指定したプログラム番号にプログラムが保存されていない
Overflow	: 指定した数値が-32768 ～ 32767 を超えている

#### ■ 利用例

プログラム番号 1 を読み込む

LOAD 1

OK

### 3.29 I2CW I2C スレーブデバイスへのデータ送信（関数）

#### ■ 書式

I2CW(デバイスアドレス,コマンドアドレス,コマンド長,データアドレス,データ長)

#### ■ 引数

デバイスアドレス : 0 ~ 127 (\$00 ~ \$7F) (7 ビット指定)  
I2C スレーブアドレスを 7 ビット形式で指定  
コマンドアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)  
送信するコマンドが格納されている SRAM 内相対アドレス  
コマンド長 : 0 ~ 32767  
送信するコマンドのバイト数  
データアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)  
送信するデータが格納されている SRAM 内相対アドレス  
データ長 : 0 ~ 32767  
送信するデータのバイト数

#### ■ 説明

I2C スレーブデバイスにデータを送信します。

送信先はデバイスアドレスにて I2C スレーブアドレスを指定します。

送信するデータは SRAM 先頭からの相対アドレスにて指定します。

送信するデータはあらかじめ、設定しておく必要があります。

コマンドとデータの区別はありません。デバイスに対しては、単純にコマンド、データの順に送信しています。

相対アドレスの指定には次の定数を利用することで有用な領域への書込み・参照が簡単に行えます。

MEM	: ユーザーワーク領域	サイズ 1024 バイト
VRAM	: 画面表示用メモリ (80×25)	サイズ 2048 バイト
VAR	: 変数領域 (A~Z)	サイズ 52 バイト
ARRAY	: 配列変数領域(@0)~@(99)	サイズ 200 バイト

I2C 通信には次の接続ピンを利用します。

PB6 : SCL (I2C クロック)

PB7 : SDA (I2C データ)

(注意) SRAM 容量を超える相対アドレス (STM32F103C8T6 の場合は 20480) への参照はエラーとなります。

#### ■ 戻り値

0 : 正常終了  
1 : 通信バッファに対してデータが長すぎる  
2 : アドレス送信に NACK が返された  
3 : データ送信に NACK が返された  
4 : その他のエラー

## ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 指定した引数の値が不当である
Out of range value	: 指定した値が有効範囲を超えている
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

## ■ 利用例

I2C EEPROM(AT24C256 スレーブアドレス \$50)にデータの読み書きを行う

```

100 POKE MEM+0,$00,$00
110 POKE MEM+2,64,65,66,67
120 R=I2CW($50,MEM,2,MEM+2,4)
130 ? "snd r=";R
140 POKE MEM+6,0,0,0,0
150 WAIT 5
160 R=I2CR($50,MEM,2,MEM+6,4)
170 ? "rcv r=";R
180 ? "rcv data:";
190 FOR I=0 TO 3
200 ? PEEK(MEM+6+I);" ";
210 NEXT I
220 ?

```

実行結果

```

run
snd r=0
rcv r=0
rcv data:64 65 66 67
OK

```

接続している I2C スレーブを調べる

(補足) I2C スレーブアドレスのみ送信し、正常終了(ACK を返した)のアドレスを調べています。

```

10 FOR I=0 TO $7F
20 C=I2CW(I,MEM,0,MEM,0)
30 IF C=0 PRINT HEX$(I,2);" ";
40 NEXT I
50 PRINT :PRINT "done."

```

実行結果

```

run
50
done.
OK

```

### 3.30 I2CR I2C スレーブデバイスからのデータ受信（関数）

#### ■ 書式

I2CR(デバイスアドレス,コマンドアドレス,コマンド長,データアドレス,データ長)

#### ■ 引数

デバイスアドレス : 0 ~ 127 (\$00 ~ \$7F) (7 ビット指定)  
I2C スレーブアドレスを 7 ビット形式で指定  
コマンドアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)  
送信するコマンドが格納されている SRAM 内相対アドレス  
コマンド長 : 0 ~ 32767  
送信するコマンドのバイト数  
受信データアドレス : 0 ~ 32767(\$0000 ~ \$1FFFF)  
受信データを格納する SRAM 内相対アドレス  
データ長 : 0 ~ 32767  
送信するデータのバイト数

#### ■ 説明

I2C スレーブデバイスからデータを受信します。

送信先はデバイスアドレスにて I2C スレーブアドレスを指定します。

受信において、コマンド等の制御データの送信が必要な場合は、コマンドアドレス、コマンド長にて送信するデータを指定します。受信のみを行う場合は、コマンドアドレス、コマンド長に 0 を設定します。送信受信するデータは SRAM 先頭からの相対アドレスにて指定します。

相対アドレスの指定には次の定数を利用することで有用な領域への書き込み・参照が簡単に行えます。

MEM	: ユーザーワーク領域	サイズ 1024 バイト
VRAM	: 画面表示用メモリ (80×25)	サイズ 2048 バイト
VAR	: 変数領域 (A~Z)	サイズ 52 バイト
ARRAY	: 配列変数領域(@0)~@(99)	サイズ 200 バイト

I2C 通信には次の接続ピンを利用します。

PB6 : SCL (I2C クロック)

PB7 : SDA (I2C データ)

(注意) SRAM 容量を超える相対アドレス (STM32F103C8T6 の場合は 20480) への参照はエラーとなります。

#### ■ 戻り値

0 : 正常終了  
1 : 通信バッファに対してデータが長すぎる  
2 : アドレス送信に NACK が返された  
3 : データ送信に NACK が返された  
4 : その他のエラー

## ■ エラーメッセージ

Syntax error	: 文法エラー、書式と異なる利用を行った
Illegal value	: 指定した引数の値が不当である
Out of range value	: 指定した値が有効範囲を超えている
Overflow	: 指定した数値が-32768 ~ 32767 を超えている

## ■ 利用例

I2C EEPROM(AT24C256 スレーブアドレス \$50)にデータの読み書きを行う

```

100 POKE MEM+0,$00,$00
110 POKE MEM+2,64,65,66,67
120 R=I2CW($50,MEM,2,MEM+2,4)
130 ? "snd r=";R
140 POKE MEM+6,0,0,0,0
150 WAIT 5
160 R=I2CR($50,MEM,2,MEM+6,4)
170 ? "rcv r=";R
180 ? "rcv data:";
190 FOR I=0 TO 3
200 ? PEEK(MEM+6+I);" ";
210 NEXT I
220 ?

```

実行結果

```

run
snd r=0
rcv r=0
rcv data:64 65 66 67
OK

```

